The Haskell School of Music

by

Paul Hudak

Yale University
Department of Computer Science

Contents

Preface **iii**

1 Computation by Calculation **1**
1.1 Computation by Calculation in Haskell 2
1.2 Expressions, Values, and Types 6
1.3 Function Types and Type Signatures 8
1.4 Abstraction, Abstraction, Abstraction 9
1.4.1 Naming 10
1.4.2 Functional Abstraction 12
1.4.3 Data Abstraction 13
1.5 Code Reuse and Modularity 16
1.6 Beware of Programming with Numbers 17

2 Simple Music **20**
2.1 Preliminaries 20
2.2 Notes and Music 22
2.3 Convenient Auxiliary Functions 24
2.4 Absolute Pitches 29

3 Polymorphic and Higher-Order Functions **31**
3.1 Polymorphic Types 31
3.2 Abstraction Over Recursive Definitions 33
3.2.1 Map is Polymorphic 34
3.2.2 Using map 35
3.3 Append 36
3.4 Fold 38
3.4.1 Haskell's Folds 39
3.4.2 Why Two Folds? 40
3.4.3 Fold for Non-empty Lists 42

3.5 A Final Example: Reverse 42
3.6 Errors 44

4 More About Higher-Order Functions 47
4.1 Currying 47
4.2 Sections 50
4.3 Anonymous Functions 51
4.4 Function Composition 53

5 More Music 56
5.1 Delay and Repeat 56
5.2 Inversion and Retrograde 57
5.3 Polyrhythms 57
5.4 Symbolic Meter Changes 58
5.5 Computing Duration 59
5.6 Super-retrograde 59
5.7 Truncating Parallel Composition 59
5.8 Trills 60
5.9 Percussion 61

6 Interpretation and Performance 64
6.1 Abstract Performance 64
6.2 Players 69
6.2.1 Examples of Player Construction 71

Preface

In 2000 I wrote a book called *The Haskell School of Expression – Learning Functional Programming through Mulitmedia*. In that book I used graphics, animation, music, and robotics as a way to motivate learning how to program, and specifically how to learn *functional programming* using Haskell, a purely functional programming language. Haskell is quite a bit different from conventional imperative or objected-oriented languages such as C, C++, Java, C#, and so on. It takes a different mind-set to program in such a language, and appeals to the mathematically inclined and to those who seek purity and elegance in their programs. Although Haskell was designed almost twenty years ago, it has only recently begun to catch on, not just because of its purity and elegance, but because with it you can solve real-world problems quickly and efficiently, and with great economy of code.

I have also had a long, informal, yet passionate interest in music, being an amateur jazz pianist and having played in several bands over the years. About ten years ago, in an effort to combine work with play, I wrote a Haskell library called *Haskore* for expressing high-level computer music concepts in a purely functional way. Indeed, three of the chapters in The Haskell School of Expression summarize the basic ideas of this work. Thus, when I recently became responsible for the Music Track in the new *Computing and the Arts* major at Yale, and became responsible for teaching not one, but two computer music courses in the new curriculum, it was natural to base the course material on Haskell. This current book is essentially a rewrite of The Haskell School of Expression with a focus entirely on music, based on and improving upon the ideas in Haskore.

Haskell was named after the logician Haskell B. Curry who, along with Alonzo Church, established the theoretical foundations of functional programming in the 1940's, when digital computers were mostly just a gleam in researchers' eyes. A curious historical fact is that Haskell Curry's father, Samuel Silas Curry, helped found and direct a school in Boston called the *School of Expression*. (This school eventually evolved into what is now

Curry College.) Since pure functional programming is centered around the notion of an *expression*, I thought that *The Haskell School of Expression* would be a good title for my first book. And it was thus quite natural to choose *The Haskell School of Music* for my second!

How To Read This Book

As mentioned earlier, there is a certain mind-set, a certain viewpoint of the world, and a certain approach to problem solving that collectively work best when programming in Haskell (this is true for any new programming paradigm). If you teach only Haskell language details to a C programmer, she is likely to write ugly, incomprehensible functional programs. But if you teach her how to think differently, how to see problems in a different light, functional solutions will come easily, and elegant Haskell programs will result. As Samuel Silas Curry once said:

> All expression comes *from within outward*, from the center to the surface, from a hidden source to outward manifestation. The study of expression as a natural process brings you into contact with cause and makes you feel the source of reality.

What is especially beautiful about this quote is that music is a kind of expression, although Curry was more likely talking about speech. In addition, as has been noted by many, music has many ties to mathematics. So for me, combining the elegant mathematical nature of Haskell with that of music is as natural as singing a nursery tune.

Using a high-level language to express musical ideas is, of course, not new. But Haskell is unique in its insistence on purity (no side effects), and this alone makes it particularly suitable for expressing musical ideas. By focusing on *what* a musical entity is rather than on *how* we should create it, we allow musical ideas to take their natural form as Haskell expressions. Haskell's many abstraction mechanisms allow us to write musical programs that are elegant, concise, yet powerful. We will consistently attempt to let the music express itself as naturally as possible, without encoding it in terms of irrelevant language details.

Of course, my ultimate goal is to teach computer music concepts. But along the way you will also learn Haskell. There is no limit to what one might wish to do with compuiter music, and therefore the better you are at programming, the more success you will have. This is why I think that many languages designed specifically for computer music—although fun to

work with, easy to use, and cute in concept—will ultimately be too limited in expressiveness.

My general approach to introducing computer music concepts is to first provide an intuitive explanation, then a mathematically rigorous definition, and finally fully executable Haskell code. In the process I will introduce Haskell features as they are needed, rather than all at once. I believe that this interleaving of concepts and applications makes the material easier to digest.

Another characteristic of my approach is that I won't hide any details—I want Haskore to be as transparent as possible! There are no magical built-in operations, no special computer music commands or values. This works out well for several reasons. First, there is in fact nothing ugly or difficult to hide—so why hide anything at all? Second, by reading the code, you will better and more quickly understand Haskell. Finally, by stepping through the design process with me, you may decide that you prefer a different approach—there is, after all, no One True Way to express computer music ideas. Indeed, I expect that this process will position you well to write rich, creative musical applications on your own.

I encourage the seasoned programmer having experience only with conventional imperative and/or object-oriented languages to read this text with an open mind. Many things will be different, and will likely feel awkward. There will be a tendency to rely on old habits when writing new programs, and to ignore suggestions about how to approach things differently. If you can manage to resist those tendencies I am confident that you will have an enjoyable learning experience. Many of those who succeed in this process find that many ideas about functional programming can be applied to imperative and object-oriented languages as well, and that their imperative coding style changes for the better.

I also ask the experienced programmer to be patient while in the earlier chapters I explain things like "syntax," "operator precedence," etc., since it is my goal that this text should be readable by someone having only modest prior programming experience. With patience the more advanced ideas will appear soon enough.

If you are a novice programmer, I suggest taking your time with the book; work through the exercises, and don't rush things. If, however, you don't fully grasp an idea, feel free to move on, but try to re-read difficult material at a later time when you have seen more examples of the concepts in action. For the most part this is a "show by example" textbook, and you should try to execute as many of the programs in this text as you can, as well as every program that you write. Learn-by-doing is the corollary to

show-by-example.

Haskell Implementations

There are several good implementations of Haskell, all available free on the Internet through the Haskell Home Page at `http://haskell.org`. One that I especially recommend is *GHC*, an easy-to-use and easy-to-install Haskell compiler and interpreter (see `http://haskell.org/ghc`). GHC runs on a variety of platforms, including PC's (Windows XP and Vista), various flavors of Unix (Linux, FreeBSD, etc.), and Mac OS X. Any text editor can be used to create the source files, but I prefer to use emacs (see `http://www.gnu.org/software/emacs`), along with its Haskell mode (see `http://www.haskell.org/haskell-mode`). All of the source code from this textbook can be found at `http://plucky.cs.yale.edu/cs431`. Feel free to email me at `paul.hudak@yale.edu` with any comments, suggestions, or questions.

Happy Haskell Music Hacking!

Paul Hudak
New Haven
September 2008

Chapter 1

Computation by Calculation

Programming, in its broadest sense, is *problem solving*. It begins when we look out into the world and see problems that we want to solve, problems that we think can and should be solved using a digital computer. Understanding the problem well is the first—and probably the most important—step in programming, since without that understanding we may find ourselves wandering aimlessly down a dead-end alley, or worse, down a fruitless alley with no end. "Solving the wrong problem" is a phrase often heard in many contexts, and we certainly don't want to be victims of that crime. So the first step in programming is answering the question, "What problem am I trying to solve?"

Once you understand the problem, then you must find a solution. This may not be easy, of course, and in fact you may discover several solutions, so we also need a way to measure success. There are various dimensions in which to do this, including correctness ("Will I get the right answer?") and efficiency ("Will I have enough resources?"). But the distinction of which solution is better is not always clear, since the number of dimensions can be large, and programs will often excel in one dimension and do poorly in others. For example, there may be one solution that is fastest, one that uses the least amount of memory, and one that is easiest to understand. Deciding which to choose can be difficult, and is one of the more interesting challenges that you will face in programming.

The last measure of success mentioned above—clarity of a program—is somewhat elusive, most difficult to measure, and, quite frankly, sometimes difficult to rationalize. But in large software systems clarity is an especially important goal, since the most important maxim about such systems is that they are never really finished! The process of continuing work on a software

system after it is delivered to users is what software engineers call *software maintenance*, and is the most expensive phase of the so-called "software life-cycle." Software maintenance includes fixing bugs in programs, as well as changing certain functionality and enhancing the system with new features in response to users' experience.

Therefore taking the time to write programs that are highly legible—easy to understand and reason about—will facilitate the software maintenance process. To complete the emphasis on this issue, it is important to realize that the person performing software maintenance is usually not the person who wrote the original program. So when you write your programs, write them as if you are writing them for someone else to see, understand, and ultimately pass judgement on!

As I work through the many musical examples in this book, I will sometimes express them in several different ways (some of which are dead-ends!), taking the time to contrast them in style, efficiency, clarity, and functionality.[1] I do this not just for pedagogical purposes. *Such reworking of programs is the norm*, and you are encouraged to get into the habit of doing so. Don't always be satisfied with your first solution to a problem, and always be prepared to go back and change—or even throw away—those parts of your program that you later discover do not fully satisfy your actual needs.

1.1 Computation by Calculation in Haskell

In this text I will use the programming language *Haskell* to address many of the issues discussed in the last section. I have tried to avoid the approach of explaining Haskell first and then giving examples second. Rather, I will walk with you, step by step, along the path of understanding a problem, understanding the solution space, and understanding how to express a particular solution in Haskell. I want you to learn how to problem solve!

Along this path I will use whatever tools are appropriate for analyzing a particular problem, very often mathematical tools that should be familiar to the average college student, indeed most to the average high-school student. As I do this I will evolve our problems toward a particular view of computation that I find especially useful: that of *computation by calculation*. You will find that such a viewpoint is not only powerful—we won't shy away from difficult problems—it is also *simple*. Haskell supports well the idea of computation by calculation. Programs in Haskell can be viewed as *functions* whose input is that of the problem being solved, and whose output is our

[1] At times I also explore different methods for proving properties of programs.

desired result; and the behavior of functions can be understood easily as computation by calculation.

An example might help to demonstrate these ideas. Suppose we want to perform an arithmetic calculation such as $3 \times (9 + 5)$. In Haskell we would write this as $3 * (9 + 5)$, since most standard computer keyboards and text editors do not recognize the special symbol $\times$. To calculate the result, we proceed as follows:

$$
\begin{aligned}
& 3 * (9 + 5) \\
\Rightarrow\ & 3 * 14 \\
\Rightarrow\ & 42
\end{aligned}
$$

It turns out that this is not the only way to compute the result, as evidenced by this alternative calculation:[2]

$$
\begin{aligned}
& 3 * (9 + 5) \\
\Rightarrow\ & 3 * 9 + 3 * 5 \\
\Rightarrow\ & 27 + 3 * 5 \\
\Rightarrow\ & 27 + 15 \\
\Rightarrow\ & 42
\end{aligned}
$$

Even though this calculation takes two extra steps, it at least gives the correct answer. Indeed, an important property of each and every program in this textbook—in fact every program that can be written in the functional language Haskell—is that it will always yield the same answer when given the same inputs, regardless of the order we choose to perform the calculations.[3] This is precisely the mathematical definition of a function: for the same inputs, it always yields the same output.

On the other hand, the first calculation above took less steps than the second, and so we say that it is more *efficient*. Efficiency in both space (amount of memory used) and time (number of steps executed) is important when searching for solutions to problems, but of course if we get the wrong answer, efficiency is a moot point. In general we will search first for any solution to a problem, and later refine it for better performance.

The above calculations are fairly trivial, of course. But we will be doing much more sophisticated operations soon enough. For starters—and to

[2] This assumes that multiplication distributes over addition in the number system being used, a point that I will return to later.

[3] As long as we don't choose a non-terminating sequence of calculations, another issue that we will return to later.

introduce the idea of a function—we could *generalize* the arithmetic operations performed in the previous example by defining a function to perform them for any numbers x, y, and z:

$$simple\ x\ y\ z = x * (y + z)$$

This equation defines *simple* as a function of three *arguments*, x, y, and z. In mathematical notation, we might see the above written slightly differently, namely:

$$simple(x, y, z) = x \times (y + z)$$

In any case, it should be clear that "*simple* 3 9 5" is the same as "$3*(9+5)$." In fact the proper way to calculate the result is:

$$\begin{aligned} &simple\ 3\ 9\ 5 \\ &\Rightarrow 3 * (9 + 5) \\ &\Rightarrow 3 * 14 \\ &\Rightarrow 42 \end{aligned}$$

The first step in this calculation is an example of *unfolding* a function definition: 3 is substituted for x, 9 for y, and 5 for z on the right-hand side of the definition of *simple*. This is an entirely mechanical process, not unlike what the computer actually does to execute the program.

When we wish to say that an expression e evaluates (via zero, one, or possibly many more steps) to the value v, we will write $e \Longrightarrow v$ (this arrow is longer than that used earlier). So we can say directly, for example, that $simple\ 3\ 9\ 5 \Longrightarrow 42$, which should be read "*simple* 3 9 5 evaluates to 42."

With *simple* now suitably defined, we can repeat the sequence of arithmetic calculations as often as we like, using different values for the arguments to *simple*. For example, $simple\ 4\ 3\ 2 \Longrightarrow 20$.

We can also use calculation to *prove properties* about programs. For example, it should be clear that for any a, b, and c, $simple\ a\ b\ c$ should yield the same result as $simple\ a\ c\ b$. For a proof of this, we calculate *symbolically*; that is, using the symbols a, b, and c rather than concrete numbers such as 3, 5, and 9:

$$\begin{aligned} &simple\ a\ b\ c \\ &\Rightarrow a * (b + c) \\ &\Rightarrow a * (c + b) \\ &\Rightarrow simple\ a\ c\ b \end{aligned}$$

I will use the same notation for these symbolic steps as for concrete ones. In particular, the arrow in the notation reflects the direction of our reasoning, and nothing more. In general, if *e1* $\Rightarrow$ *e2*, then it's also true that *e2* $\Rightarrow$ *e1*.

I will also refer to these symbolic steps as "calculations," even though the computer will not typically perform them when executing a program (although it might perform them *before* a program is run if it thinks that it might make the program run faster). The second step in the calculation above relies on the commutativity of addition (namely that, for any numbers x and y, $x + y = y + x$). The third step is the reverse of an unfold step, and is appropriately called a *fold* calculation. It would be particularly strange if a computer performed this step while executing a program, since it does not seem to be headed toward a final answer. But for proving properties about programs, such "backward reasoning" is quite important.

When I wish to make the justification for each step clearer, whether symbolic or concrete, I will present a calculation with more detail, as in:

simple a b c
$\Rightarrow$ { *unfold* }
$a * (b + c)$
$\Rightarrow$ { *commutativity* }
$a * (c + b)$
$\Rightarrow$ { *fold* }
simple a c b

In most cases, however, this will not be necessary.

Proving properties of programs is another theme that will be repeated often in this text. As the world begins to rely more and more on computers to accomplish not just ordinary tasks such as writing term papers and sending email, but also life-critical tasks such as controlling medical procedures and guiding spacecraft, then the correctness of the programs that we write gains in importance. Proving complex properties of large, complex programs is not easy—and rarely if ever done in practice—but that should not deter us from proving simpler properties of the whole system, or complex properties of parts of the system, since such proofs may uncover errors, and if not, at least help us to gain confidence in our effort.

If you are someone who is already an experienced programmer, the idea of computing *everything* by calculation may seem odd at best, and naive at worst. How does one write to a file, draw a picture, play a sound, or respond to mouse-clicks? If you are wondering about these things, I hope that you have patience reading the early chapters, and that you find delight in reading the later chapters where the full power of this approach begins to shine. I

will avoid, however, most comparisons between Haskell and conventional programming languages such as C, C++, Java, or even Scheme or ML (two "almost functional" languages).

In many ways this first chapter is the most difficult chapter in the entire text, since it contains the highest density of new concepts. If you have trouble with some of the ideas here, keep in mind that we will return to almost every idea at later points in the text. And don't hesistate to return to this chapter later to re-read difficult sections; they will likely be much easier to grasp at that time.

Exercise 1.1 Write out all of the steps in the calculation of the value of

simple (*simple* 2 3 4) 5 6

Exercise 1.2 Prove by calculation that *simple* $(a - b)$ a $b \Longrightarrow a^2 - b^2$.

> **Details:** In the remainder of the text the need will often arise to explain some aspect of Haskell in more detail, without distracting too much from the primary line of discourse. In those circumstance I will off-set the comments and proceed them with the word "Details," such as is done with this paragraph.

1.2 Expressions, Values, and Types

In Haskell, the entities that we perform calculations on are called *expressions*, and the entities that result from a calculation—i.e. "the answers"—are called *values*. It is helpful to think of a value just as an expression on which no more calculation can be carried out.

Examples of expressions include *atomic* (meaning, indivisible) values such as the integer 42 and the character `'a'`, which are examples of two *primitive* atomic values. In the next Chapter we will will also see examples of *user-defined* atomic values, such as the pitch classes C, Cs, Df, etc. (denoting the musical notes C, C♯, D♭, etc.).

In addition, there are *structured* (meaning, made from smaller pieces) expressions such as the list $[C, Cs, Df]$ and the pair (`'b'`, 4) (lists and pairs are different in a subtle way, to be described later). Each of these structured expressions is also a value, since by themselves there is no calculation that can be carried out. As another example, $1 + 2$ is an expression, and one step of calculation yields the expression 3, which is a value, since no more calculations can be performed.

Sometimes, however, an expression has only a never-ending sequence of calculations. For example, if x is defined as:

$$x = x + 1$$

then here's what happens when we try to calculate the value of x:

$$\begin{aligned} & x \\ \Rightarrow\ & x + 1 \\ \Rightarrow\ & (x + 1) + 1 \\ \Rightarrow\ & ((x + 1) + 1) + 1 \\ \Rightarrow\ & (((x + 1) + 1) + 1) + 1 \\ & \ldots \end{aligned}$$

This is clearly a never-ending sequence of steps, in which case we say that the expression does not terminate, or is *non-terminating*. In such cases the symbol *bottom*, pronounced "bottom," is used to denote the value of the expression.

Every expression (and therefore every value) also has an associated *type*. You can think of types as sets of expressions (or values), in which members of the same set have much in common. Examples include the atomic types *Integer* (the set of all fixed-precision integers) and *Char* (the set of all characters), as well as the structured types $[Integer]$ and $[PitchClass]$ (the set of all lists of integers and pitch classes, respectively) and $(Char, Integer)$ (the set of all character/integer pairs). The association of an expression or value with its type is very important, and there is a special way of expressing it in Haskell. Using the examples of values and types above, we write:

$$\begin{aligned} 42 &:: Integer \\ \texttt{'a'} &:: Char \\ [C, Cs, Df] &:: [PitchClass] \\ (\texttt{'b'}, 4) &:: (Char, Integer) \end{aligned}$$

> **Details:** Literal characters are written enclosed in single forward quotes, as in `'a'`, `'A'`, `'b'`, `','`, `'!'`, `' '` (a space), etc. (There are some exceptions, however; see the Haskell Report for details.)

The "::" should be read "has type," as in "42 has type *Integer*."

> **Details:** Note that the names of specific types are capitalized, such as *Integer* and *Char*, but the names of values are not, such as *simple* and x. This is not just a convention: it is required when programming in Haskell.

> In addition, the case of the other characters matters, too. For example, *test*, *teSt*, and *tEST* are all distinct names for values, as are *Test*, *TeST*, and *TEST* for types.

Haskell's *type system* ensures that Haskell programs are *well-typed*; that is, that the programmer has not mismatched types in some way. For example, it does not make much sense to add together two characters, so the expression `'a'+'b'` is *ill-typed*. The best news is that Haskell's type system will tell you if your program is well-typed *before you run it*. This is a big advantage, since most programming errors are manifested as typing errors.

1.3 Function Types and Type Signatures

What should the type of a function be? It seems that it should at least convey the fact that a function takes values of one type—*T1*, say—as input, and returns values of (possibly) some other type—*T2*, say—as output. In Haskell this is written $T1 \rightarrow T2$, and we say that such a function "maps values of type *T1* to values of type *T2*." If there is more than one argument, the notation is extended with more arrows. For example, if our intent is that the function *simple* defined in the previous section has type $Integer \rightarrow Integer \rightarrow Integer \rightarrow Integer$, we can declare this fact by including a *type signature* with the definition of *simple*:

$$simple :: Integer \rightarrow Integer \rightarrow Integer \rightarrow Integer$$
$$simple\ x\ y\ z = x * (y + z)$$

> **Details:** When you write Haskell programs using a typical text editor, you will not see nice fonts and arrows as in $Integer \rightarrow Integer$. Rather, you will have to type `Integer -> Integer`.

Haskell's type system also ensures that user-supplied type signatures such as this one are correct. Actually, Haskell's type system is powerful enough to allow us to avoid writing any type signatures at all, in which case we say that the type system *infers* the correct types for us.[4] Nevertheless, judicious placement of type signatures, as we did for *simple*, is a good habit, since type signatures are an effective form of documentation and help bring programming errors to light. Also, in almost every example in this text, I

[4]There are a few exceptions this rule, and in the case of *simple* the inferred type is actually a bit more general than that written above. Both of these points will be returned to later.

will make a habit of first talking about the types of expressions and functions as a way to better understand the problem at hand, organize our thoughts, and lay down the first ideas of a solution.

The normal use of a function is referred to as *function application*. For example, *simple* 3 9 5 is the application of the function *simple* to the arguments 3, 9, and 5.

> **Details:** Some functions, such as $(+)$, are applied using what is known as *infix syntax*; that is, the function is written between the two arguments rather than in front of them (compare $x + y$ to $f\ x\ y$). Infix functions are often called *operators*, and are distinguished by the fact that they do not contain any numbers or letters of the alphabet. Thus ^! and $*\#$: are infix operators, whereas $thisIsAFunction$ and $f9g$ are not (but are still valid names for functions or other values). The only exception to this is that the symbol ′ is considered to be alphanumeric; thus f' and $one's$ are valid names, but not operators.
>
> In Haskell, when referring to an operator as a value, it is enclosed in parentheses, such as when declaring its type, as in:
>
> $$(+) :: Integer \to Integer \to Integer$$
>
> Also, when trying to understand an expression such as $f\ x + g\ y$, there is a simple rule to remember: function application always has "higher precedence" than operator application, so that $f\ x + g\ y$ is the same as $(f\ x) + (g\ y)$.
>
> Despite all of these syntactic differences, however, operators are still just functions.

Exercise 1.3 Identify the well-typed expressions in the following and, for each, give its proper type:

$[(2, 3), (4, 5)]$
$[Cs, 42]$
$(Df, -42)$
simple `'a' 'b' 'c'`
$(simple\ 1\ 2\ 3, simple)$

1.4 Abstraction, Abstraction, Abstraction

The title of this section is the answer to the question: "What are the three most important ideas in programming?" Well, perhaps this is an overstatement, but I hope that I've gotten your attention, at least. Webster defines the verb "abstract" as follows:

> **abstract**, *vt* (1) remove, separate (2) to consider apart from application to a particular instance.

In programming we do this when we see a repeating pattern of some sort, and wish to "separate" that pattern from the "particular instances" in which it appears. Let's refer to this process as the *abstraction principle*, and see how it might manifest itself in problem solving.

1.4.1 Naming

One of the most basic ideas in programming—for that matter, in every day life—is to *name* things. For example, we may wish to give a name to the value of π, since it is inconvenient to retype (or remember) the value of π beyond a small number of digits. In mathematics the greek letter π in fact *is* the name for this value, but unfortunately we don't have the luxury of using greek letters on standard computer keyboards and text editors. So in Haskell we write:

$$pi :: Float$$
$$pi = 3.1415927$$

to associate the name *pi* with the number 3.1415927. The type signature in the first line declares *pi* to be a *floating-point number*, which mathematically—and in Haskell—is distinct from an integer.[5] Now we can use the name *pi* in expressions whenever we want; it is an abstract representation, if you will, of the number 3.1415927. Furthermore, if we ever have a need to change a named value (which hopefully won't ever happen for *pi*, but could certainly happen for other values), we would only have to change it in one place, instead of in the possibly large number of places where it is used.

Suppose now that we are working on a problem whose solution requires writing some expression more than once. For example, we might find ourselves computing something such as:

$$x :: Float$$
$$x = f\ (a - b + 2) + g\ y\ (a - b + 2)$$

The first line declares x to be a floating-point number, while the second is an equation that defines the value of x. Note on the right-hand side of this equation that the expression $a - b + 2$ is repeated—it has two instances—and thus, applying the abstraction principle, we wish to separate it from

[5] I will have more to say about floating-point numbers later in this chapter.

these instances. We already know how to do this—it's called *naming*—so we might choose to rewrite the single equation above as two:

$$c = a - b + 2$$
$$x = f\ c + g\ y\ c$$

If, however, the definition of c is not intended for use elsewhere in the program, then it is advantageous to "hide" the definition of c within the definition of x. This will avoid cluttering up the namespace, and prevents c from clashing with some other value named c. To achieve this, we simply use a **let** expression:

$$x = \textbf{let}\ c = a - b + 2$$
$$\quad \textbf{in}\ f\ c + g\ y\ c$$

A **let** expression restricts the *visibility* of the names that it creates to the internal workings of the **let** expression itself. For example, if we write:

$$c = 42$$
$$x = \textbf{let}\ c = a - b + 2$$
$$\quad \textbf{in}\ f\ c + g\ y\ c$$

then there is no conflict of names—the "outer" c is completely different from the "inner" one enclosed in the **let** expression. Think of the inner c as analogous to the first name of someone in your household. If your brother's name is "John" he will not be confused with John Thompson who lives down the street when you say, "John spilled the milk."

> **Details:** An equation such as $c = 42$ is called a *binding*. A simple rule to remember when programming in Haskell is never to give more than one binding for the same name in a context where the names can be confused, whether at the top level of your program or nestled within a **let** expression. For example, this is not allowed:
>
> $$a = 42$$
> $$a = 43$$
>
> nor is this:
>
> $$a = 42$$
> $$b = 43$$
> $$a = 44$$

So you can see that naming—using either top-level equations or equations within a **let** expression—is an example of the abstraction principle in action.

1.4.2 Functional Abstraction

[I should replace the following with something musical, but for now it will have to do as is.]

Let's now consider a more complex example. Suppose we are computing the sum of the areas of three circles with radii *r1*, *r2*, and *r3*, as expressed by:

```
totalArea :: Float
totalArea = pi * r1^2 + pi * r2^2 + pi * r3^2
```

Details: (^) is Haskell's integer exponentiation operator. In mathematics we would write $\pi \times r^2$ or just πr^2 instead of $pi * r\hat{}2$.

Although there isn't an obvious repeating expression here as there was in the last example, there is a repeating *pattern of operations*. Namely, the operations that square some given quantity—in this case the radius—and then multiply the result by π. To abstract a sequence of operations such as this, we use a *function*—which we will give the name *circleArea*—that takes the "given quantity"—the radius—as an argument. There are three instances of the pattern, each of which we can expect to replace with a call to *circleArea*. This leads to:

```
circleArea :: Float → Float
circleArea r = pi * r^2

totalArea = circleArea r1 + circleArea r2 + circleArea r3
```

Using the idea of unfolding described earlier, it is easy to verify that this definition is equivalent to the previous one.

This application of the abstraction principle is sometimes called *functional abstraction*, since the sequence of operations is abstracted as a function, in this case *circleArea*. Actually, it can be seen as a generalization of the previous kind of abstraction: *naming*. That is, *circleArea r1* is just a name for *pi*r1*^2, *circleArea r2* for *pi*r2*^2, and *circleArea r3* for *pi*r3*^2. Or in other words, a named quantity such as *c* or *pi* defined previously can be thought of as a function with no arguments.

Note that *circleArea* takes a radius (a floating-point number) as an argument and returns the area (also a floating-point number) as a result, as reflected in its type signature.

The definition of *circleArea* could also be hidden within *totalArea* using a **let** expression as we did in the previous example:

$$\begin{aligned} totalArea = \; & \textbf{let } circleArea\ r = pi * r\hat{}2 \\ & \textbf{in } circleArea\ r1 + circleArea\ r2 + circleArea\ r3 \end{aligned}$$

On the other hand, it is more likely that computing the area of a circle will be useful elsewhere in the program, so leaving the definition at the top level is probably preferable in this case.

1.4.3 Data Abstraction

The value of *totalArea* is the sum of the areas of three circles. But what if in another situation we must add the areas of five circles, or in other situations even more? In situations where the number of things is not certain, it is useful to represent them in a *list* whose length is arbitrary. So imagine that we are given an entire list of circle areas, whose length isn't known at the time we write the program. What now?

I will define a function *listSum* to add the elements of a list. Before doing so, however, there is a bit more to say about lists.

Lists are an example of a *data structure*, and when their use is motivated by the abstraction principle, I will say that we are applying *data abstraction*. Earlier we saw the example $[1, 2, 3]$ as a list of integers, whose type is thus $[Integer]$. A list with *no* elements is—not surprisingly—written *[]*, and pronounced "nil." To add a single element x to the front of a list xs, we write $x : xs$. (Note the naming convention used here; xs is the plural of x, and should be read that way.) In fact, the list $[1, 2, 3]$ is equivalent to $1 : (2 : (3 : [\,]))$, which can also be written $1 : 2 : 3 : [\,]$ since the infix operator *(:)* is "right associative."

> **Details:** In mathematics we rarely worry about whether the notation $a + b + c$ stands for $(a + b) + c$ (in which case $+$ would be "left associative") or $a + (b + c)$ (in which case $+$ would "right associative"). This is because in situations where the parentheses are left out it's usually the case that the operator is *mathematically* associative, meaning that it doesn't matter which interpretation we choose. If the interpretation *does* matter, mathemeticians will include parentheses to make it clear. Furthermore, in mathematics there is an implicit assumption that some operators have higher *precedence* than others; for example, $2 \times a + b$ is interpreted as $(2 \times a) + b$, not $2 \times (a + b)$.
>
> In most programming languages, including Haskell, each operator is defined as having some precedence level and to be either left or right associative. For arithmetic operators, mathematical convention is usually followed; for example, $2 * a + b$ is interpreted as $(2 * a) + b$ in Haskell. The predefined list-forming operator (:) is defined to be right associative. Just

as in mathematics, this associativey can be over-ridden by using parentheses: thus $(a : b) : c$ is a valid Haskell expression (assuming that it is well-typed), and is very different from $a : b : c$. I will explain later how to specify the associativity and precedence of new operators that we define.

Examples of pre-defined functions defined on lists in Haskell include *head* and *tail*, which return the "head" and "tail" of a list, respectively. That is, $head\ (x : xs) \Rightarrow x$ and $tail\ (x : xs) \Rightarrow xs$ (we will define these two functions formally in Section 3.1). Another example is the function $(+\!\!+)$ which *concatenates*, or *appends*, together its two list arguments. For example, $[1, 2, 3] +\!\!+ [4, 5, 6] \Rightarrow [1, 2, 3, 4, 5, 6]$ ($(+\!\!+)$ will be defined in Section **??**).

Returning to the problem of defining a function to add the elements of a list, let's first express what its type should be:

$$listSum :: [Float] \rightarrow Float$$

Now we must define its behavior appropriately. Often in solving problems such as this it is helpful to consider, one by one, all possible cases that could arise. To compute the sum of the elements of a list, what might the list look like? The list could be empty, in which case the sum is surely 0. So we write:

$$listSum\ [\,] = 0$$

The other possibility is that the list *isn't* empty—i.e. it contains at least one element—in which case the sum is the first number plus the sum of the remainder of the list. So we write:

$$listSum\ (x : xs) = x + listSum\ xs$$

Combining these two equations with the type signature brings us to the complete definition of the function *listSum*:

$$\begin{aligned} &listSum :: [Float] \rightarrow Float \\ &listSum\ [\,] = 0 \\ &listSum\ (x : xs) = x + listSum\ xs \end{aligned}$$

Details: Although intuitive, this example highlights an important aspect of Haskell: *pattern matching*. The left-hand sides of the equations contain *patterns* such as $[\,]$ and $x : xs$. When a function is applied, these patterns are *matched* against the argument values in a fairly intuitive way ($[\,]$ only matches the empty list, and $x : xs$ will successfully match any list with at

least one element, while naming the first element x and the rest of the list xs). If the match succeeds, the right-hand side is evaluated and returned as the result of the application. If it fails, the next equation is tried, and if all equations fail, an error results. All of the equations that define a particular function must appear together, one after the other.

Defining functions by pattern matching is quite common in Haskell, and you should eventually become familiar with the various kinds of patterns that are allowed; see Appendix **??** for a concise summary.

This is called a *recursive* function definition since *listSum* "refers to itself" on the right-hand side of the second equation. Recursion is a very powerful technique that you will see used many times in this text. It is also an example of a general problem-solving technique where a large problem is broken down into many simpler but similar problems; solving these simpler problems one-by-one leads to a solution to the larger problem.

Here is an example of *listSum* in action:

$$
\begin{aligned}
& \mathit{listSum}\ [1,2,3] \\
\Rightarrow\ & \mathit{listSum}\ (1:(2:(3:[\,]))) \\
\Rightarrow\ & 1 + \mathit{listSum}\ (2:(3:[\,])) \\
\Rightarrow\ & 1 + (2 + \mathit{listSum}\ (3:[\,])) \\
\Rightarrow\ & 1 + (2 + (3 + \mathit{listSum}\ [\,])) \\
\Rightarrow\ & 1 + (2 + (3 + 0)) \\
\Rightarrow\ & 1 + (2 + 3) \\
\Rightarrow\ & 1 + 5 \\
\Rightarrow\ & 6
\end{aligned}
$$

The first step above is not really a calculation, but rather a rewriting of the list syntax. The remaining calculations consist of four unfold steps followed by three integer additions.

Given this definition of *listSum* we can rewrite the definition of *totalArea* as:

$$\mathit{totalArea} = \mathit{listSum}\ [\mathit{circleArea}\ \mathit{r1}, \mathit{circleArea}\ \mathit{r2}, \mathit{circleArea}\ \mathit{r3}]$$

This may not seem like much of an improvement, but if we were adding many such circle areas in some other context, it would be. Indeed, lists are arguably the most commonly used structured data type in Haskell. In the next chapter we will see a more convincing example of the use of lists; namely, to represent the vertices that make up a polygon. Since a polygon can have an arbitrary number of vertices, using a data structure such as a list seems like just the right approach.

In any case, how do we know that this version of *totalArea* behaves the same as the original one? By calculation, of course:

$$
\begin{array}{l}
\mathit{listSum}\ [\mathit{circleArea}\ \mathit{r1}, \mathit{circleArea}\ \mathit{r2}, \mathit{circleArea}\ \mathit{r3}] \\
\Longrightarrow \{\ \mathit{unfold}\ \mathit{listSum}\ (\mathit{four\ succesive\ times})\ \} \\
\mathit{circleArea}\ \mathit{r1} + \mathit{circleArea}\ \mathit{r2} + \mathit{circleArea}\ \mathit{r3} + 0 \\
\Longrightarrow \{\ \mathit{unfold}\ \mathit{circleArea}\ (\mathit{three\ places})\ \} \\
\mathit{pi} * \mathit{r1}\hat{}2 + \mathit{pi} * \mathit{r2}\hat{}2 + \mathit{pi} * \mathit{r3}\hat{}2 + 0 \\
\Rightarrow \{\ \mathit{simple\ arithmetic}\ \} \\
\mathit{pi} * \mathit{r1}\hat{}2 + \mathit{pi} * \mathit{r2}\hat{}2 + \mathit{pi} * \mathit{r3}\hat{}2
\end{array}
$$

1.5 Code Reuse and Modularity

There doesn't seem to be much repetition in our last definition for *totalArea*, so perhaps we're done. In fact, let's pause for a moment and consider how much progress we've made. We started with the definition:

$$\mathit{totalArea} = \mathit{pi} * \mathit{r1}\hat{}2 + \mathit{pi} * \mathit{r2}\hat{}2 + \mathit{pi} * \mathit{r3}\hat{}2$$

and ended with:

$$\mathit{totalArea} = \mathit{listSum}\ [\mathit{circleArea}\ \mathit{r1}, \mathit{circleArea}\ \mathit{r2}, \mathit{circleArea}\ \mathit{r3}]$$

But additionally, we have introduced definitions for the auxiliary functions *circleArea* and *listSum*. In terms of size, our final program is actually larger than what we began with! So have we actually improved things?

From the standpoint of "removing repeating patterns," we certainly have, and we could argue that the resulting program is easier to understand as a result. But there is more. Now that we have defined auxiliary functions such as *circleArea* and *listSum*, we can *reuse* them in other contexts. Being able to reuse code is also called *modularity*, since the reused components are like little modules, or bricks, that can form the foundation of many applications.[6] We've already talked about reusing *circleArea*; and *listSum* is surely reusable: imagine a list of grocery item prices, or class sizes, or city populations, for each of which we must compute the total. In later chapters you will learn other concepts—most notably higher-order functions and polymorphism—that will substantially increase your ability to reuse code.

[6] "Code reuse" and "modularity" are important software engineering principles.

1.6 Beware of Programming with Numbers

In mathematics there are many different kinds of number systems. For example, there are integers, natural numbers (i.e. non-negative integers), real numbers, rational numbers, and complex numbers. These number systems possess many useful properties, such as the fact that multiplication and addition are commutative, and that multiplication distributes over addition. You have undoubtedly learned many of these properties in your studies, and have used them often in algebra, geometry, trigonometry, physics, etc.

Unfortunately, each of these number systems places great demands on computer systems. In particular, a number can in general require an *arbitrary amount of memory* to represent it—even an infinite amount! Clearly, for example, we cannot represent an irrational number such as π exactly; the best we can do is approximate it, or possibly write a program that computes it to whatever (finite) precision that we need in a given application. But even integers (and therefore rational numbers) present problems, since any given integer can be arbitrarily large.

Most programming languages do not deal with these problems very well. In fact, most programming languages do not have exact forms of any of these number systems. Haskell does slightly better than most, in that it has exact forms of integers (the type *Integer*) as well as rational numbers (the type *Rational*, defined in the Ratio Library). But in Haskell and most other languages there is no exact form of real numbers, for example, which are instead approximated by *floating-point numbers* with either single-word precision (*Float* in Haskell) or double-word precision (*Double*). What's worse, the behavior of arithmetic operations on floating-point numbers can vary somewhat depending on what CPU is being used, although hardware standardization in recent years has reduced the degree of this problem.

The bottom line is that, as simple as they may seem, great care must be taken when programming with numbers. Many computer errors, some quite serious and renowned, were rooted in numerical incongruities. The field of mathematics known as *numerical analysis* is concerned precisely with these problems, and programming with floating-point numbers in sophisticated applications often requires a good understanding of numerical analysis to devise proper algorithms and write correct programs.

As a simple example of this problem, consider the distributive law, expressed here as a calculation in Haskell and used earlier in this chapter in calculations involving the function *simple*:

$$a * (b + c) \Rightarrow a * b + a * c$$

For most floating-point numbers, this law is perfectly valid. For example, in the GHC implementation of Haskell, the expressions $pi * (3 + 4) :: Float$ and $pi * 3 + pi * 4 :: Float$ both yield the same result: 21.99115. But funny things can happen when the magnitude of $b + c$ differs significantly from the magnitude of either b or c. For example, the following two calculations are from GHC:

$$5 * (-0.123456 + 0.123457) :: Float \Rightarrow 4.991889\text{e} - 6$$
$$5 * (-0.123456) + 5 * (0.123457) :: Float \Rightarrow 5.00679\text{e} - 6$$

Although the error here is small, its very existence is worrisome, and in certain situations it could be disastrous. I will not discuss the nature of floating-point numbers much further in this text, but just remember that they are *approximations* to the real numbers. If real-number accuracy is important to your application, further study of the nature of floating-point numbers is probably warranted.

On the other hand, the distributive law (and many others) is valid in Haskell for the exact data types *Integer* and *Ratio Integer* (i.e. rationals). However, another problem arises: although the representation of an *Integer* in Haskell is not normally something that we are concerned about, it should be clear that the representation must be allowed to grow to an arbitrary size. For example, Haskell has no problem with the following number:

$$veryBigNumber :: Integer$$
$$veryBigNumber = 432083457203485932198765123721340 59$$

and such numbers can be added, multiplied, etc. without any loss of accuracy. However, such numbers cannot fit into a single word of computer memory, most of which are limited to 32 bits. Worse, since the computer system does not know ahead of time exactly how many words will be required, it must devise a dynamic scheme to allow just the right number of words to be used in each case. The overhead of implementing this idea unfortunately causes programs to run slower.

For this reason, Haskell provides another integer data type called *Int* which has maximum and minimum values that depend on the word-size of the CPU being used. In other words, every value of type *Int* fits into one word of memory, and the primitive machine instructions for integers can be used to manipulate them very efficiently.[7] Unfortunately, this means

[7] The Haskell Report requires that every implementation support *Int*s in the range -2^{29} to $2^{29} - 1$, inclusive. The GHC implementation running on a Pentium processor, for example, supports the range -2^{31} to $2^{31} - 1$.

that *overflow* or *underflow* errors could occur when an *Int* value exceeds either the maximum or minumum values. However, most implementations of Haskell (as well as most other languages) do not even tell you when this happens. For example, in GHC, the following *Int* value:

$$i :: Int$$
$$i = 1234567890$$

works just fine, but if you multiply it by two, GHC returns the value -1825831516! This is because twice i exceeds the maximum allowed value, so the resulting bits become nonsensical,[8] and are interpreted in this case as a negative number of the given magnitude.

This is alarming! Indeed, why should anyone ever use *Int* when *Integer* is available? The answer, as mentioned earlier, is efficiency, but clearly care should be taken when making this choice. If you are indexing into a list, for example, and you are confident that you are not performing index calculations that might result in the above kind of error, then *Int* should work just fine, since a list longer than 2^{31} will not fit into memory anyway! But if you are calculating the number of microseconds in some large time interval, or counting the number of people living on earth, then *Integer* would most likely be a better choice. Choose your number data types wisely!

In this text I will use the data types *Integer*, *Int*, *Float*, *Double* and *Rational* for a variety of different applications; for a discussion of the other number types, consult the Haskell Report. As I use these data types, I will do so without much discussion—this is not, after all, a book on numerical analysis—but I will issue a warning whenever reasoning about floating-point numbers, for example, in a way that might not be technically sound.

[8] Actually, they are perfectly sensible in the following way: the 32-bit binary representation of i is 01001001100101100000001011010010, and twice that is 10010011001011000000010110100100. But the latter number is seen as negative because the 32nd bit (the highest-order bit on the CPU on which this was run) is a one, which means it is a negative number in "twos-complement" representation. The twos-complement of this number is in turn 01101100110100111111101001011100, whose decimal representation is 1825831516.

Chapter 2

Simple Music

```
module Haskore.Music where
import Ix
import Ratio
infixr 5 :+:, :=:
```

In the previous chapter we introduced some of the fundamental ideas of functional programming in Haskell. In this chapter we begin to develop some *musical* ideas as well. As we do so, more Haskell features will be introduced.

2.1 Preliminaries

Sometimes it is useful to use a built-in Haskell data type to directly represent some concept of interest. For example, we may wish to use *Int* to represent octaves, where by convention octave 4 corresponds to the octave containing middle C on the piano. We can express this in Haskell using a *type synonym*:

```
type Octave = Int
```

A type synonym does not create a new data type—it just gives a new name to an existing type. Type synonyms can be defined not just for atomic types such as *Int*, but also for structured types such as pairs. For example, in music theory a pitch is normally defined as a pair, a pitch class and an octave. Assuming the existence of a data type called *PitchClass*, we can write the following type synonym:

```
type Pitch = (PitchClass, Octave)
```

For example, "concert A," i.e. A above middle C (sometimes written A4) corresponds to the pitch $(A, 4)$. For convenience we could define a Haskell variable with that value as follows:

```
a4 :: Pitch
a4 = (A, 4)      -- concert A
```

> **Details:** This example also demonstrates the use of program *comments*. Any text to the right of " -- " till the end of the line is considered to be a comment, and is effectively ignored. Haskell also permits *nested* comments that have the form {- this is a comment -} and can appear anywhere in a program.

Another useful musical concept is *duration*. Rather than use either integers or floating-point numbers, we will use *rational* numbers to denote duration:

```
type Dur = Rational
```

Rational is the data type of rational numbers expressed as ratios of *Integer*s in Haskell. Rational numbers in Haskell are written in the form $n \,\%\, m$, where n is the numerator, and d is the denominator. Thus we can define, for example, a quarter note as follows:

```
qn :: Dur
qn = 1 % 4       -- quarter note
```

So far so good. But what about *PitchClass*? We might try to use integers to represent pitch classes as well, but this is not very elegant—ideally we would like to write something that looks more like the conventional pitch class names C, C♯, D♭, D, etc. The solution is to use an *algebraic data type* in Haskell:

```
data PitchClass = Cf | C | Cs | Df | D | Ds | Ef | E | Es | Ff | F | Fs
                | Gf | G | Gs | Af | A | As | Bf | B | Bs
    deriving (Eq, Ord, Ix, Show, Read)
```

Ignoring the line beginning with "**deriving**" for the moment, this data type declaration simply enumerates the 21 pitch class names (three for each of the note names A through G). Note that enharmonics (such as G♯ and A♭) are listed separately, which may be important in certain applications.

> **Details:** All constructors in a **data** declaration must be capitalized. In this way they are syntactically distinguished from ordinary values. This distinction is useful since only constructors can be used in the pattern matching that is part of a function definition, as will be described shortly.

Keep in mind that *PitchClass* is a completely new, user-defined data type that is not equal to any other.

2.2 Notes and Music

We can of course define other data types for other purposes. For example, we will want to define the notion of a *note* (the pairing of a pitch with a duration), and a *rest*. Both of these can be thought of as *primitive* musical values, and thus we write:

```
data Prim = Note Dur Pitch
          | Rest Dur
    deriving (Show, Eq, Ord)
```

For example, *Note qn a4* is concert A played as a quarter note, and *Rest* 1 is a whole-note rest.

This definition is not completely satisfactory, however, because we may wish to attach other information to a note, such as its loudness, or some other annotation or articulation. Furthermore, the pitch itself may actually be a percussive sound, having no true pitch at all. To fix this we will introduce an important concept in Haskell, namely *polymorphism*—the ability to parameterize over types. Instead of fixing the type of the pitch of a note, we will leave it unspecified through the use of a *type variable*, as follows:

```
data Primitive a = Note Dur a
                 | Rest Dur
    deriving (Show, Eq, Ord)
```

Note the type variable a, which is used as an argument to *Primitive*, and then used in the body of the declaration—just like a variable in a function. *Primitive Pitch* is now the same as (or, technically, is now *isomorphic to*) the type *Prim*. Indeed, instead of defining *Prim* as above, we could now use a type synonym instead:

```
type Prim = Primitive Pitch
```

But *Primitive* is more flexible than *Prim*, since, for example, we could add loudness by pairing loudness with pitch, as in *Primitive* (*Pitch*, *Loudness*). We will see more concrete instances of this idea later.

So far we only have a way to express primitive notes and rests—how do we combine many notes and rests into a larger composition? To achieve this we will define another polymorphic data type, perhaps the most important data type used in this book, which defines the fundamental structure of a musical entity:

```
data Music a = Primitive (Primitive a)      -- primitive value
             | Music a :+: Music a          -- sequential composition
             | Music a :=: Music a          -- parallel composition
             | Modify Control (Music a)     -- modifier
  deriving (Show, Eq, Ord)
```

Details: The first line here looks odd: the name *Primitive* appears twice. The first occurence, however, is the name of a new *constructor* in the *Music* data type, whereas the second is the name of the existing *data type* defined above. Haskell allows using the same name to define a constructor and a data type, since they can never be confused: the context in which they are used will always be sufficient to distinguish them.

Also note the use of *infix constructors* (:+:) and (:=:). Infix constructors are just like infix operators in Haskell, but they must begin with a colon. This distinction exists to make it easier to pattern match, and is analogous to the distinction between ordinary names (which must begin with a lowercase character) and constructor names (which must begin with an uppercase character).

It is convenient to represent these musical ideas as a recursive datatype because we wish to not only construct musical values, but also take them apart, analyze their structure, print them in a structure-preserving way, interpret them for performance purposes, etc. We will see many examples of these kinds of processes shortly.

This data type declaration essentially says that a value of type *Music a* has one of four possible forms:

- *Primitive p*, where p is a primitive value of type *Primitive a*, for some type *a*. For example:

```
ma4 :: Music Pitch
ma4 = Primitive (Note qn a4)
```

is the musical value corresponding to a quarter-note rendition of concert A.

- *m1* :+: *m2* is the *sequential* composition of *m1* and *m2*; i.e. *m1* and *m2* are played in sequence.
- *m1* :=: *m2* is the *parallel* composition of *m1* and *m2*; i.e. *m1* and *m2* are played simultaneously.
- *Modify cntrl m* is an "annotated" version of *m* in which the control parameter *cntrl* specifies some way in which *m* is to be modified.

Details: Note that *Music a* is defined in terms of *Music a*, and thus we say that is a *recursive* data type. It is also often called an *inductive* data type, since it is, in essence, an inductive definition of an infinite number of values, each of which can be arbitrarily complex.

The *Control* data type is defined as follows:

```
data Control =
        Tempo Rational        -- scale the tempo
      | Transpose AbsPitch    -- transposition
      | Instrument InstrumentName    -- intrument label
      | Phrase [PhraseAttribute]     -- phrase attributes
      | Player PlayerName     -- player label
    deriving (Show, Eq, Ord)

type PlayerName = String
```

It allows one to annotate a *Music* value with a tempo change, a transposition, a phrase attribute, a player name, or an instrument. Instrument names are borrowed from the General MIDI standard, and are captured as an algebraic data type in Figure **??**. Phrase attributes and the concept of a "player" are closely related, but a full explanation is deferred until Chapter 6.

2.3 Convenient Auxiliary Functions

For convenient we define a number of functions to make it easier to write certain kinds of musical values. For starters, we define:

```
note d p = Primitive (Note d p)
rest d = Primitive (Rest d)
```

```
data InstrumentName
  = AcousticGrandPiano | BrightAcousticPiano | ElectricGrandPiano
  | HonkyTonkPiano | RhodesPiano | ChorusedPiano
  | Harpsichord | Clavinet | Celesta | Glockenspiel | MusicBox
  | Vibraphone | Marimba | Xylophone | TubularBells
  | Dulcimer | HammondOrgan | PercussiveOrgan
  | RockOrgan | ChurchOrgan | ReedOrgan
  | Accordion | Harmonica | TangoAccordion
  | AcousticGuitarNylon | AcousticGuitarSteel | ElectricGuitarJazz
  | ElectricGuitarClean | ElectricGuitarMuted | OverdrivenGuitar
  | DistortionGuitar | GuitarHarmonics | AcousticBass
  | ElectricBassFingered | ElectricBassPicked | FretlessBass
  | SlapBass1 | SlapBass2 | SynthBass1 | SynthBass2
  | Violin | Viola | Cello | Contrabass | TremoloStrings
  | PizzicatoStrings | OrchestralHarp | Timpani
  | StringEnsemble1 | StringEnsemble2 | SynthStrings1
  | SynthStrings2 | ChoirAahs | VoiceOohs | SynthVoice
  | OrchestraHit | Trumpet | Trombone | Tuba
  | MutedTrumpet | FrenchHorn | BrassSection | SynthBrass1
  | SynthBrass2 | SopranoSax | AltoSax | TenorSax
  | BaritoneSax | Oboe | Bassoon | EnglishHorn | Clarinet
  | Piccolo | Flute | Recorder | PanFlute | BlownBottle
  | Shakuhachi | Whistle | Ocarina | Lead1Square
  | Lead2Sawtooth | Lead3Calliope | Lead4Chiff
  | Lead5Charang | Lead6Voice | Lead7Fifths
  | Lead8BassLead | Pad1NewAge | Pad2Warm
  | Pad3Polysynth | Pad4Choir | Pad5Bowed
  | Pad6Metallic | Pad7Halo | Pad8Sweep
  | FX1Train | FX2Soundtrack | FX3Crystal
  | FX4Atmosphere | FX5Brightness | FX6Goblins
  | FX7Echoes | FX8SciFi | Sitar | Banjo | Shamisen
  | Koto | Kalimba | Bagpipe | Fiddle | Shanai
  | TinkleBell | Agogo | SteelDrums | Woodblock | TaikoDrum
  | MelodicDrum | SynthDrum | ReverseCymbal
  | GuitarFretNoise | BreathNoise | Seashore
  | BirdTweet | TelephoneRing | Helicopter
  | Applause | Gunshot | Percussion
  | Custom String
  deriving (Show, Eq, Ord)
```

Figure 2.1: General MIDI Instrument Names

```
cf, c, cs, df, d, ds, ef, e, es, ff, f, fs, gf, g, gs, af, a, as, bf, b, bs::
    Octave → Dur → Music Pitch

cf o d = note d (Cf, o)
c o d = note d (C, o)
cs o d = note d (Cs, o)
df o d = note d (Df, o)
d o d = note d (D, o)
ds o d = note d (Ds, o)
ef o d = note d (Ef, o)
e o d = note d (E, o)
es o d = note d (Es, o)
ff o d = note d (Ff, o)
f o d = note d (F, o)
fs o d = note d (Fs, o)
gf o d = note d (Gf, o)
g o d = note d (G, o)
gs o d = note d (Gs, o)
af o d = note d (Af, o)
a o d = note d (A, o)
as o d = note d (As, o)
bf o d = note d (Bf, o)
b o d = note d (B, o)
bs o d = note d (Bs, o)
```

Figure 2.2: Convenient note names.

```
tempo r m = Modify (Tempo r) m
transpose i m = Modify (Transpose i) m
instrument i m = Modify (Instrument i) m
phrase pa m = Modify (Phrase pa) m
player pn m = Modify (Player pn) m
```

We can also create simple names for familiar notes, durations, and rests, as shown in Figures 2.2 and 2.3. Despite the large number of them, these names are sufficiently "unusual" that name clashes are unlikely.

As a simple example, here is a ii-V-I chord progression in C major:

```
t251 :: Music Pitch
t251 = let dMinor = d 3 wn :=: f 3 1 :=: a 3 wn
           gMajor = g 3 wn :=: b 3 1 :=: d 4 wn
           cMajor = c 3 bn :=: e 3 2 :=: g 3 bn
       in dMinor :+: gMajor :+: cMajor
```

```
bn, wn, hn, qn, en, sn, tn, sfn :: Dur
dwn, dhn, dqn, den, dsn, dtn :: Dur
ddhn, ddqn, dden :: Dur

bnr, wnr, hnr, qnr, enr, snr, tnr :: Music Pitch
dwnr, dhnr, dqnr, denr, dsnr, dtnr :: Music Pitch
ddhnr, ddqnr, ddenr :: Music Pitch

bn = 2; bnr = rest bn       -- brevis rest
wn = 1; wnr = rest wn       -- whole note rest
hn = 1 % 2; hnr = rest hn       -- half note rest
qn = 1 % 4; qnr = rest qn       -- quarter note rest
en = 1 % 8; enr = rest en       -- eight note rest
sn = 1 % 16; snr = rest sn      -- sixteenth note rest
tn = 1 % 32; tnr = rest tn      -- thirty-second note rest
sfn = 1 % 64; sfnr = rest sfn       -- sixty-fourth note rest

dwn = 3 % 2; dwnr = rest dwn      -- dotted whole note rest
dhn = 3 % 4; dhnr = rest dhn      -- dotted half note rest
dqn = 3 % 8; dqnr = rest dqn      -- dotted quarter note rest
den = 3 % 16; denr = rest den      -- dotted eighth note rest
dsn = 3 % 32; dsnr = rest dsn      -- dotted sixteenth note rest
dtn = 3 % 64; dtnr = rest dtn      -- dotted thirty-second note rest

ddhn = 7 % 8; ddhnr = rest ddhn       -- double-dotted half note rest
ddqn = 7 % 16; ddqnr = rest ddqn      -- double-dotted quarter note rest
dden = 7 % 32; ddenr = rest dden      -- double-dotted eighth note rest
```

Figure 2.3: Convenient rest names.

Details: Note that more than one equation is allowed in a **let** expression. The first characters of each equation, however, must line up vertically, and if an equation takes more than one line then the subsequent lines must be to the right of the first characters. For example, this is legal:

```
let a = aLongName
        + anEvenLongerName
    b = 56
in...
```

but neither of these are:

```
let a = aLongName
  +anEvenLongerName
    b = 56
in...

let a = aLongName
        + anEvenLongerName
  b = 56
in...
```

(The second line of the first example is too far to the left, as is the third line in the second example.)

Although this rule, called the *layout rule*, may seem a bit *ad hoc*, it avoids having to use special syntax to denote the end of one equation and the beginning of the next (such as a semicolon), thus enhancing readability. In practice, use of layout is rather intuitive. Just remember two things:

First, the first character following either **where** or **let** (and a few other keywords that we will see later) is what determines the starting column for the set of equations being written. Thus we can begin the equations on the same line as the keyword, the next line, or whatever.

Second, just be sure that the starting column is further to the right than the starting column associated with any immediately surrounding clause (otherwise it would be ambiguous). The "termination" of an equation happens when something appears at or to the left of the starting column associated with that equation.

In order to play this simple example, we can import the *play* function from Hasore's MIDI library, and simply type:

```
play t251
```

at the GHC command line. Default instruments and tempos are used to then play the resulting composition.

2.4 Absolute Pitches

Treating pitches simply as integers is useful in many settings, so let's use a type synonym to introduce a concept of "absolute pitch:"

type $AbsPitch = Int$

The absolute pitch of a (relative) pitch can be defined mathematically as 12 times the octave, plus the index of the pitch class. We can express this in Haskell as follows:

$$absPitch :: Pitch \rightarrow AbsPitch$$
$$absPitch\ (pc, oct) = 12 * oct + pcToInt\ pc$$

Details: Note the use of *pattern-matching* to match the argument of $absPitch$ to a pair.

$pcToInt$ is simply a function that converts a particular pitch class to an index, easily expressed as:

$$pcToInt :: PitchClass \rightarrow Int$$
$$pcToInt\ Cf = -1$$
$$pcToInt\ C = 0$$
$$pcToInt\ Cs = 1$$
$$pcToInt\ Df = 1$$
$$pcToInt\ D = 2$$
$$pcToInt\ Ds = 3$$
$$pcToInt\ Ef = 3$$
$$pcToInt\ E = 4$$
$$pcToInt\ Es = 5$$
$$pcToInt\ Ff = 4$$
$$pcToInt\ F = 5$$
$$pcToInt\ Fs = 6$$
$$pcToInt\ Gf = 6$$
$$pcToInt\ G = 7$$
$$pcToInt\ Gs = 8$$
$$pcToInt\ Af = 8$$
$$pcToInt\ A = 9$$
$$pcToInt\ As = 10$$
$$pcToInt\ Bf = 10$$
$$pcToInt\ B = 11$$
$$pcToInt\ Bs = 12$$

Converting an absolute pitch to a pitch is a bit more tricky, because of enharmonic equivalences. For example, the absolute pitch 15 might correspond to either $(Ds, 1)$ or $(Ef, 1)$. We take the approach of always returning a sharp in such ambiguous cases:

$$\begin{aligned} &pitch :: AbsPitch \to Pitch \\ &pitch\ ap = ([\,C, Cs, D, Ds, E, F, Fs, G, Gs, A, As, B\,] \mathbin{!!} mod\ ap\ 12, \\ &\qquad\qquad quot\ ap\ 12) \end{aligned}$$

Details: (!!) is Haskell's zero-based list-indexing function; $list \mathbin{!!} n$ returns the $(n+1)$th element in $list$. $mod\ x\ n$ is the value of x modulo n; and $quot\ x\ n$ is the integer quotient of x divided by n.

We can also define a function *trans*, which transposes pitches:

$$\begin{aligned} &trans :: Int \to Pitch \to Pitch \\ &trans\ i\ p = pitch\ (absPitch\ p + i) \end{aligned}$$

Exercise 2.1 *Show that* $abspitch \circ pitch = id$*, and, up to enharmonic equivalences,* $pitch \circ abspitch = id$*.*

Exercise 2.2 *Show that* $trans\ i\ (trans\ j\ p) = trans\ (i + j)\ p$*.*

Chapter 3

Polymorphic and Higher-Order Functions

In the last chapter we learned a little about polymorphic data types. In this chapter we will also learn about *polymorphic functions*, which are essentially functions defined over polymorphic data types. The already familiar *list* is the most common example of a polymorphic data type, and I will discuss it at length in this chapter. Although lists have no direct musical connection, they are perhaps the most commonly used data type in Haskell, and have many applications in computer music programming.

We will also learn about *higher-order functions*, which are functions that take one or more functions as arguments or return a function as a result (functions can also be placed in data structures, making the data constructors higher-order too). Together, polymorphic and higher-order functions substantially increase our expressive power and our ability to reuse code. We will see that both of these new ideas naturally follow the foundations that we have already built.

(A more detailed discussion of pre-defined polymorphic functions that operate on lists can be found in Chapter **??**.)

3.1 Polymorphic Types

In previous chapters we saw examples of lists of several different kinds of elements—integers, characters, pitch classes, and so on—and you can well imagine situations requiring lists of other element types as well. Sometimes, however, we don't wish to be so particular about the precise type of the elements. For example, suppose we want to define a function *length* that de-

termines the number of elements in a list. We don't really care whether the list contains integers, pitch classes, or even other lists—we imagine computing the length in exactly the same way in each case. The obvious definition is:

$$
\begin{aligned}
&\mathit{length}\ [\,] = 0\\
&\mathit{length}\ (x : \mathit{xs}) = 1 + \mathit{length}\ \mathit{xs}
\end{aligned}
$$

This recursive definition is self-explanatory. We can read the equations as saying: "The length of the empty list is 0, and the length of a list whose first element is x and remainder is xs is 1 plus the length of xs."

But what should the type of *length* be? Intuitively, what we'd like to say is that, for *any* type a, the type of *length* is $[a] \to \mathit{Integer}$. In Haskell we write this simply as:

$$\mathit{length} :: [a] \to \mathit{Integer}$$

> **Details:** Generic names for types, such as a above, are called *type variables*, and are uncapitalized to distinguish them from specific types such as $\mathit{Integer}$.

So *length* can be applied to a list containing elements of *any* type. For example:

$$
\begin{aligned}
&\mathit{length}\ [1,2,3] &&\Longrightarrow\ 3\\
&\mathit{length}\ [C, \mathit{Cs}, \mathit{Df}] &&\Longrightarrow\ 3\\
&\mathit{length}\ [[1],[],[2,3,4]] &&\Longrightarrow\ 3
\end{aligned}
$$

Note that the type of the argument to *length* in the last example is $[[\mathit{Integer}]]$; that is, a list of lists of integers.

Here are two other examples of polymorphic list functions, which happen to be pre-defined in Haskell:

$$
\begin{aligned}
&\mathit{head} :: [a] \to a\\
&\mathit{head}\ (x : \_) = x
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{tail} :: [a] \to [a]\\
&\mathit{tail}\ (\_ : \mathit{xs}) = \mathit{xs}
\end{aligned}
$$

> **Details:** The _ on the left-hand side of these equations is called a *wildcard* pattern. It matches any value, and binds no variables. It is useful as a way of documenting the fact that we do not care about the value in that part of the pattern.

These two functions take the "head" and "tail," respectively, of any non-empty list:

$$
\begin{aligned}
&\mathit{head}\ [1,2,3] \Rightarrow 1\\
&\mathit{head}\ [\texttt{'a'},\texttt{'b'},\texttt{'c'}] \Rightarrow \texttt{'a'}\\
&\mathit{tail}\ [1,2,3] \Rightarrow [2,3]\\
&\mathit{tail}\ [\texttt{'a'},\texttt{'b'},\texttt{'c'}] \Rightarrow [\texttt{'b'},\texttt{'c'}]
\end{aligned}
$$

Functions such as *length*, *head*, and *tail* are said to be *polymorphic* (*poly* means *many* and *morphic* refers to the structure, or *form*, of objects). Polymorphic functions arise naturally when defining functions on lists and other polymorphic data types, including the *Music* data type defined in the last chapter.

3.2 Abstraction Over Recursive Definitions

Suppose we have a list of pitches, and we wish to convert each of them to an absolute pitch. We might write a function:

$$
\begin{aligned}
&\mathit{toAbsPitches} :: [\mathit{Pitch}] \to [\mathit{AbsPitch}]\\
&\mathit{toAbsPitches}\ [\,] = [\,]\\
&\mathit{toAbsPitches}\ (p : ps) = \mathit{absPitch}\ p : \mathit{toAbsPitches}\ ps
\end{aligned}
$$

We might also want to convert a list of absolute pitches to a list of pitches:

$$
\begin{aligned}
&\mathit{toPitches} :: [\mathit{AbsPitch}] \to [\mathit{Pitch}]\\
&\mathit{toPitches}\ [\,] = [\,]\\
&\mathit{toPitches}\ (a : as) = \mathit{pitch}\ a : \mathit{toPitches}\ as
\end{aligned}
$$

These two functions are different, but share something in common: there is a repeating pattern of operations. But the pattern is not quite like any of the examples that we studied earlier, and therefore it is unclear how to apply the abstraction principle. What distinguishes this situation is that there is a repeating pattern of *recursion*.

In discerning the nature of a repeating pattern it's sometimes helpful to identify those things that *aren't* repeating—i.e. those things that are *changing*—since these will be the sources of *parameterization*: those values that must be passed as arguments to the abstracted function. In the case above, these changing values are the functions *absPitch* and *pitch*; let's consider them instances of a new name, f. If we then simply rewrite either of the above functions as a new function—let's call it *map*—that takes an extra argument f, we arrive at:

$$map\ f\ [\,] = [\,]$$
$$map\ f\ (x : xs) = f\ x : map\ f\ xs$$

With this definition of *map*, we can now redefine *toAbsPitches* and *toPitches* as:

$$toAbsPitches :: [Pitch] \to [AbsPitch]$$
$$toAbsPitches\ ps = map\ absPitch\ ps$$

$$toPitches :: [AbsPitch] \to [Pitch]$$
$$toPitches\ as = map\ pitch\ as$$

Note that these definitions are non-recursive; the common pattern of recursion has been abstracted away and isolated in the definition of *map*. They are also very succinct; so much so, that it seems unnecessary to create new names for these functions at all! One of the powers of higher-order functions is that they permit concise yet easy-to-understand definitions such as this, and you will see many similar examples throughout the remainder of the text.

A proof that the new versions of these two functions are equivalent to the old ones can be done via calculation, but requires a proof technique called *induction*, because of the recursive nature of the original function definitions. We will discuss inductive proofs in detail, including these two examples, in Chapter **??**.

3.2.1 Map is Polymorphic

What should the type of *map* be? Let's look first at its use in *toAbsPitches*: it takes the function $absPitch :: Pitch \to AbsPitch$ as its first argument, a list of *Pitch*s as its second argument, and it returns a list of *AbsPitch*s as its result. So its type must be:

$$map :: (Pitch \to AbsPitch) \to [Pitch] \to [AbsPitch]$$

Yet a similar analysis of its use in *toPitches* reveals that *map*'s type should be:

$$map :: (AbsPitch \to Pitch) \to [AbsPitch] \to [Pitch]$$

This apparent anomaly can be resolved by noting that *map*, like *length*, *head* and *tail*, does not really care what its list element types are, *as long as its functional argument can be applied to them.* Indeed, *map* is *polymorphic*, and its most general type is:

$$map :: (a \to b) \to [a] \to [b]$$

This can be read: "*map* is a function that takes a function from any type a to any type b, and a list of a's, and returns a list of b's." The correspondence between the two a's and between the two b's is important: a function that converts *Int*'s to *Char*'s, for example, cannot be mapped over a list of *Char*'s. It is easy to see that in the case of *toAbsPitches*, a is instantiated as *Pitch* and b as *AbsPitch*, whereas in *toPitches*, a and b are instantiated as *AbsPitch* and *Pitch*, respectively.

> **Details:** In Chapter 1 we mentioned that every expression in Haskell has an associated type. But with polymorphism, you might wonder if there is just one type for every expression. For example, *map* could have any of these types:
>
> $(a \to b) \to [a] \to [b]$
> $(Integer \to b) \to [Integer] \to [b]$
> $(a \to Float) \to [a] \to [Float]$
> $(Char \to Char) \to [Char] \to [Char]$
>
> and so on, depending on how it will be used. However, notice that the first of these types is in some fundamental sense more general than the other three. In fact, every expression in Haskell has a unique type known as its *principal type*: the least general type that captures all valid uses of the expression. The first type above is the principal type of *map*, since it captures all valid uses of *map*, yet is less general than, for example, the type $a \to b \to c$. As another example, the principal type of *head* is $[a] \to a$; the types $[b] \to a$, $b \to a$, or even a are too general, whereas something like $[Integer] \to Integer$ is too specific.[1]

3.2.2 Using map

Now that we can picture *map* as a polymorphic function, it is useful to look back on some of the examples we have worked through to see if there are any situations where *map* might have been useful. For example, recall from Section 1.4.3 the definition of *totalArea*:

$$totalArea = listSum\ [circleArea\ r1, circleArea\ r2, circleArea\ r3]$$

It should be clear that this can be rewritten as:

[1] The existence of unique principal types is the hallmark feature of the *Hindley-Milner type system* [**?**, **?**] that forms the basis of the type systems of Haskell, ML [**?**] and many other functional languages [**?**].

$$totalArea = listSum\ (map\ circleArea\ [r1, r2, r3])$$

A simple calculation is all that is needed to show that these are the same:

$$\begin{aligned}
&map\ circleArea\ [r1, r2, r3] \\
&\Rightarrow circleArea\ r1 : map\ circleArea\ [r2, r3] \\
&\Rightarrow circleArea\ r1 : circleArea\ r2 : map\ circleArea\ [r3] \\
&\Rightarrow circleArea\ r1 : circleArea\ r2 : circleArea\ r3 : map\ circleArea\ [] \\
&\Rightarrow circleArea\ r1 : circleArea\ r2 : circleArea\ r3 : [] \\
&\Rightarrow [circleArea\ r1, circleArea\ r2, circleArea\ r3]
\end{aligned}$$

For an interesting musical example, let's generate a whole-tone scale starting at a given pitch:

$$\begin{aligned}
&wts :: Pitch \rightarrow [Music\ Pitch] \\
&wts\ p = \textbf{let}\ ap = absPitch\ p \\
&\qquad\qquad f\ ap = note\ qn\ (pitch\ ap) \\
&\qquad\quad \textbf{in}\ map\ f\ [mc, mc + 2\,..\,mc + 12]
\end{aligned}$$

Details: A list $[a, b\,..\,c]$ is called an *arithmetic sequence*, and is special syntax for the list $[a, a + d, a + 2 * d, ..., c]$ where $d = b - a$.

3.3 Append

Let's now consider the problem of *concatenating* or *appending* two lists together; that is, creating a third list that consists of all of the elements from the first list followed by all of the elements of the second. Once again the type of list elements does not matter, so we will define this as a polymorphic infix operator $(+\!\!+)$:

$$(+\!\!+) :: [a] \rightarrow [a] \rightarrow [a]$$

For example, here are two uses of $(+\!\!+)$ on different types:

$$\begin{aligned}
&[1, 2, 3] +\!\!+ [4, 5, 6] \Longrightarrow [1, 2, 3, 4, 5, 6] \\
&[C, E, G] +\!\!+ [D, F, A] \Longrightarrow [C, E, G, D, F, A]
\end{aligned}$$

As usual, we can approach this problem by considering the various possibilities that could arise as input. But in the case of $(+\!\!+)$ we are given *two* inputs—so which do we consider first? In general this is not an easy question, but in the case of $(+\!\!+)$ we can get a hint about what to do by noting that the result contains firstly all of the elements from the first list. So let's consider the first list first: it could be empty, or non-empty. If it is empty the answer is easy:

$$[\,] +\!\!+ ys = ys$$

and if it is not empty the answer is also straightforward:

$$(x : xs) +\!\!+ ys = x : (xs +\!\!+ ys)$$

Note the recursive use of $(+\!\!+)$. Our full definition is thus:

$$(+\!\!+) :: [a] \to [a] \to [a]$$
$$[\,] +\!\!+ ys = ys$$
$$(x : xs) +\!\!+ ys = x : (xs +\!\!+ ys)$$

The Efficiency and Fixity of Append In Chapter **??** we will prove the following simple property about $(+\!\!+)$:

$$(xs +\!\!+ ys) +\!\!+ zs = xs +\!\!+ (ys +\!\!+ zs)$$

That is, $(+\!\!+)$ is *associative*.

But what about the efficiency of the left-hand and right-hand sides of this equation? It is easy to see via calculation that appending two lists together takes a number of steps proportional to the length of the first list (indeed the second list is not evaluated at all). For example:

$$[1, 2, 3] +\!\!+ xs$$
$$\Rightarrow 1 : ([2, 3] +\!\!+ xs)$$
$$\Rightarrow 1 : 2 : ([3] +\!\!+ xs)$$
$$\Rightarrow 1 : 2 : 3 : ([\,] +\!\!+ xs)$$
$$\Rightarrow 1 : 2 : 3 : xs$$

Therefore the evaluation of $xs +\!\!+ (ys +\!\!+ zs)$ takes a number of steps proportional to the length of xs plus the length of ys. But what about $(xs +\!\!+ ys) +\!\!+ zs$? The leftmost append will take a number of steps proportional to the length of xs, but then the rightmost append will require a number of steps proportional to the length of xs plus the length of ys, for a total cost of:

$$2 * \mathit{length}\ xs + \mathit{length}\ ys$$

Thus $xs +\!\!+ (ys +\!\!+ zs)$ is more efficient than $(xs +\!\!+ ys) +\!\!+ zs$. This is why the Standard Prelude defines the fixity of $(+\!\!+)$ as:

infixr $5+\!\!+$

In other words, if you just write $xs +\!\!+ ys +\!\!+ zs$, you will get the most efficient association, namely the right association $xs +\!\!+ (ys +\!\!+ zs)$. In the next section I will demonstrate a more dramatic example of this property.

3.4 Fold

Suppose we wish to take a list of notes (each of type *Music*) and convert them into a *line*, or *melody*. We can define a recursive function to do this:

$$
\begin{aligned}
&line :: [Music\ a] \to Music\ a \\
&line\ [\,] = rest\ 0 \\
&line\ (m : ms) = m :+: line\ ms
\end{aligned}
$$

In a different situation we might wish to compute the highest pitch in a list of pitches:

$$
\begin{aligned}
&maxPitch :: [Pitch] \to Pitch \\
&maxPitch\ [\,] = 0 \\
&maxPitch\ (p : ps) = p\ !!!\ maxPitch\ ps
\end{aligned}
$$

where !!! is defined as:

$$p1\ !!!\ p2 = \textbf{if}\ absPitch\ p1 > absPitch\ p2\ \textbf{then}\ p1\ \textbf{else}\ p2$$

Once again we have a situation where several definitions share something in common—a repeating recursive pattern. Using the process that we used to discover *map*, let's first identify those things that are changing. There are two pairs: the *Primitive* (*Rest* 0) and 0 values (for which we'll use the generic name *init*, for "initial value"), and the (:+:) and (!!!) operators (for which we'll use the generic name *op*, for "operator"). If we now rewrite either of the above functions as a new function—lets call it *fold*—that takes extra arguments *op* and *init*, we arrive at:[2]

$$
\begin{aligned}
&fold\ op\ init\ [\,] = init \\
&fold\ op\ init\ (x : xs) = x\ `op`\ fold\ op\ init\ xs
\end{aligned}
$$

> **Details:** Any normal binary function name can be used as an infix operator by enclosing it in backquotes; $x\ `f`\ y$ is equivalent to $f\ x\ y$. Using infix application here for *op* better reflects the structure of the repeating pattern that we are abstracting.

With this definition of *fold* we can now rewrite the definitions of *line* and *maxPitch* as:

[2] The use of the name "*fold*" for this function is historical, and has little to do with the use of "fold" and "unfold" to describe steps in a calculation.

$$line :: [Music] \to Music$$
$$line\ ms = fold\ (:+:)\ (Primitive\ (Rest\ 0))\ ms$$

$$maxPitch :: [Pitch] \to Pitch$$
$$maxPitch\ ps = fold\ (!!!)\ 0\ ps$$

Details: Just as we can turn a function into an operator by enclosing it in backquotes, we can turn an operator into a function by enclosing it in parentheses. This is required in order to pass an operator as a value to another function, as in the examples above. (If we wrote $fold\ !!!\ 0\ ps$ instead of $fold\ (!!!)\ 0\ ps$ it would look like we were trying to compare $fold$ to $0\ ps$, which is nonsensical and ill-typed.)

In Chapter **??** we will use induction to prove that these new definitions are equivalent to the old ones.

As another example, recall the definition of *listSum* from Section 1.4.3:

$$listSum :: [Float] \to Float$$
$$listSum\ [\,] = 0$$
$$listSum\ (x : xs) = x + listSum\ xs$$

We can now rewrite this more succinctly using *fold*:

$$listSum :: [Float] \to Float$$
$$listSum\ xs = fold\ (+)\ 0\ xs$$

fold, like *map*, is a highly useful—reusable—function, as we will see through several other examples later in the text. Indeed, it too is polymorphic, for note that it does not depend on the type of the list elements. Its most general type—somewhat trickier than that for *map*—is:

$$fold :: (a \to b \to b) \to b \to [a] \to b$$

This allows us to use *fold* whenever we need to "collapse" a list of elements using a binary (i.e. two-argument) operator.

3.4.1 Haskell's Folds

Haskell actually defines two versions of *fold* in the Standard Prelude. The first is called *foldr* ("fold-from-the-right") which is defined the same as our *fold*:

$$
\begin{aligned}
&\mathit{foldr} :: (a \to b \to b) \to b \to [a] \to b \\
&\mathit{foldr}\ \mathit{op}\ \mathit{init}\ [\,] = \mathit{init} \\
&\mathit{foldr}\ \mathit{op}\ \mathit{init}\ (x : \mathit{xs}) = x\ `\mathit{op}`\ \mathit{foldr}\ \mathit{op}\ \mathit{init}\ \mathit{xs}
\end{aligned}
$$

A good way to think about *foldr* is that it replaces all occurences of the list operator (:) with its first argument (a function), and replaces [ ] with its second argument. In other words:

$$
\begin{aligned}
&\mathit{foldr}\ \mathit{op}\ \mathit{init}\ (\mathit{x1} : \mathit{x2} : ... : \mathit{xn} : [\,]) \\
&\Longrightarrow \mathit{x1}\ `\mathit{op}`\ (\mathit{x2}\ `\mathit{op}`\ (...(\mathit{xn}\ `\mathit{op}`\ \mathit{init})...))
\end{aligned}
$$

This might help you to understand the type of *foldr* better, and also explains its name: the list is "folded from the right." Stated another way, for any list *xs*, the following always holds:[3]

$$\mathit{foldr}\ (:)\ [\,]\ \mathit{xs} \Longrightarrow \mathit{xs}$$

Haskell's second version of *fold* is called *foldl*:

$$
\begin{aligned}
&\mathit{foldl} :: (b \to a \to b) \to b \to [a] \to b \\
&\mathit{foldl}\ \mathit{op}\ \mathit{init}\ [\,] = \mathit{init} \\
&\mathit{foldl}\ \mathit{op}\ \mathit{init}\ (x : \mathit{xs}) = \mathit{foldl}\ \mathit{op}\ (\mathit{init}\ `\mathit{op}`\ x)\ \mathit{xs}
\end{aligned}
$$

A good way to think about *foldl* is to imagine "folding the list from the left:"

$$
\begin{aligned}
&\mathit{foldl}\ \mathit{op}\ \mathit{init}\ (\mathit{x1} : \mathit{x2} : ... : \mathit{xn} : [\,]) \\
&\Longrightarrow (...((\mathit{init}\ `\mathit{op}`\ \mathit{x1})\ `\mathit{op}`\ \mathit{x2})...)\ `\mathit{op}`\ \mathit{xn}
\end{aligned}
$$

3.4.2 Why Two Folds?

Note that if we had used *foldl* instead of *foldr* in the definitions given earlier then not much would change; *foldr* and *foldl* would give the same result. Indeed, judging from their types, it looks like the only difference between *foldr* and *foldl* is that the operator takes its arguments in a different order.

So why does Haskell define two versions of *fold*? It turns out that there are situations where using one is more efficient, and possibly "more defined," than the other. (By more defined, I mean that the function terminates on more values of its input domain.)

Probably the simplest example of this is a generalization of the associativity of (++) discussed in the last section. Suppose that we wish to collapse a list of lists into one list. The Standard Prelude defines the polymorphic function *concat* for this purpose:

[3] We will formally prove this in Chapter **??**.

$$concat :: [[a]] \to [a]$$
$$concat\ xss = foldr\ (+\!\!+)\ [\,]\ xss$$

For example:

$$concat\ [[1],[3,4],[\,],[5,6]]$$
$$\Rightarrow [1,2,3,4,5,6]$$

More importantly, from the earlier discussion it should be clear that this property holds:

$$concat\ [xs1, xs2, ..., xsn]$$
$$\Rightarrow foldr\ (+\!\!+)\ [\,]\ [xs1, xs2, ..., xsn]$$
$$\Rightarrow xs1 +\!\!+ (xs2 +\!\!+ (...(xn +\!\!+ [\,]))...)$$

The total cost of this computation is proportional to the sum of the lengths of all of the lists. If each list has the same length *len*, then this cost is $n * len$.

On the other hand, if we had defined *concat* this way:

$$slowConcat\ xss = foldl\ (+\!\!+)\ [\,]\ xss$$

then we have:

$$slowConcat\ [xs1, xs2, ..., xsn]$$
$$\Rightarrow foldl\ (+\!\!+)\ [\,]\ [xs1, xs2, ..., xsn]$$
$$\Rightarrow (...(([\,] +\!\!+ x1) +\!\!+ x2)...) +\!\!+ xn$$

If each list has the same length *len*, then the cost of this computation will be:

$$len + (len + len) + (len + len + len) + ... + (n-1) * len$$
$$\Rightarrow n * (n-1) * len$$

which is considerably worse than $n * len$. Thus the choice of *foldr* in the definition of *concat* is quite important.

Similar examples can be given to demonstrate that *foldl* is sometimes more efficient than *foldr*. On the other hand, in many cases the choice does not matter at all (consider, for example, (+)). The moral of all this is that care must be taken in the choice between *foldr* and *foldl* if efficiency is a concern.

3.4.3 Fold for Non-empty Lists

One might argue that both *line* and *maxPitch* should not be well defined on an empty list, and for that purpose the Standard Prelude provides functions *foldr1* and *foldl1*, which return an error if applied to an empty list. Our preferred definitions for *line* and *maxPitch*, as well as a function *chord* that is similar to *line* except that it does parallel composition, are:

$$
\begin{aligned}
&line, chord :: [Music] \to Music \\
&line\ ms = foldr1\ (:+:)\ ms \\
&chord\ ms = foldr1\ (:=:)\ ms \\
\\
&maxPitch :: [Pitch] \to Pitch \\
&maxPitch\ ps = foldr1\ (!!!)\ ps
\end{aligned}
$$

3.5 A Final Example: Reverse

As a final example of a useful list function, consider the problem of *reversing* a list, which we will capture in a function called *reverse*. For example, *reverse* $[1, 2, 3]$ is $[3, 2, 1]$. Thus *reverse* takes a single list argument, whose possibilities are the normal ones for a list: it is either empty, or it is not. And so we write:

$$
\begin{aligned}
&reverse :: [a] \to [a] \\
&reverse\ [\,] = [\,] \\
&reverse\ (x : xs) = reverse\ xs +\!\!+ [x]
\end{aligned}
$$

This, in fact, is a perfectly good definition for *reverse*—it is certainly clear—except for one small problem: it is terribly inefficient! To see why, first note that the number of steps needed to compute $xs +\!\!+ ys$ is proportional to the length of xs. Now suppose that the list argument to *reverse* has length n. The recursive call to *reverse* will return a list of length $n - 1$, which is the first argument to $(+\!\!+)$. Thus the cost to reverse a list of length of n will be proportional to $n - 1$ plus the cost to reverse a list of length $n - 1$. So the total cost is proportional to $(n - 1) + (n - 2) + \cdots + 1 = n(n - 1)/2$, which in turn is proportional to the square of n.

Can we do better than this? Yes we can.

There is another algorithm for reversing a list, which goes something like this: take the first element, and put it at the front of an empty auxiliary list; then take the next element and add it to the front of the auxiliary list (thus the auxiliary list now consists of the first two elements in the original

list, but in reverse order); then do this again and again until you reach the end of the original list. At that point the auxiliary list will be the reverse of the original one.

This algorithm can be expressed recursively, but the auxiliary list implies that we need a function that takes *two* arguments—the original list and the auxiliary one—yet *reverse* only takes one. So we create an auxiliary function *rev*:

$$
\begin{aligned}
&\mathit{reverse}\ \mathit{xs} = \mathit{rev}\ [\,]\ \mathit{xs}\\
&\quad \textbf{where}\ \mathit{rev}\ \mathit{acc}\ [\,] = \mathit{acc}\\
&\qquad\qquad \mathit{rev}\ \mathit{acc}\ (x : \mathit{xs}) = \mathit{rev}\ (x : \mathit{acc})\ \mathit{xs}
\end{aligned}
$$

The auxiliary list is the first argument to *rev*, and is called *acc* since it behaves as an "accumulator" of the intermediate results. Note how it is returned as the final result once the end of the original list is reached.

A little thought should convince the reader that this function does not have the quadratic (n^2) behavior of the first algorithm, and indeed can be shown to execute a number of steps that is directly proportional to the length of the list, which we can hardly expect to improve upon.

But now, compare the definition of *rev* with the definition of *foldl*:

$$
\begin{aligned}
&\mathit{foldl}\ \mathit{op}\ \mathit{init}\ [\,] = \mathit{init}\\
&\mathit{foldl}\ \mathit{op}\ \mathit{init}\ (x : \mathit{xs}) = \mathit{foldl}\ \mathit{op}\ (\mathit{init}\ `\mathit{op}`\ x)\ \mathit{xs}
\end{aligned}
$$

They are somewhat similar. In fact, suppose we were to slightly rewrite *rev*, yielding:

$$
\begin{aligned}
&\mathit{rev}\ \mathit{op}\ \mathit{acc}\ [\,] = \mathit{acc}\\
&\mathit{rev}\ \mathit{op}\ \mathit{acc}\ (x : \mathit{xs}) = \mathit{rev}\ \mathit{op}\ (\mathit{acc}\ `\mathit{op}`\ x)\ \mathit{xs}
\end{aligned}
$$

Now *rev* looks exactly like *foldl*, and the question becomes whether or not there is a function that can be substituted for *op* that would make the latter definition of *rev* equivalent to the former one. Indeed there is:

$$\mathit{revOp}\ a\ b = b : a$$

For note that:

$$\mathit{acc}\ `\mathit{revOp}`\ x \Rightarrow \mathit{revOp}\ \mathit{acc}\ x \Rightarrow x : \mathit{acc}$$

So *reverse* can be rewritten as:

$$
\begin{aligned}
&\mathit{reverse}\ \mathit{xs} = \mathit{rev}\ \mathit{revOp}\ [\,]\ \mathit{xs}\\
&\quad \textbf{where}\ \mathit{rev}\ \mathit{op}\ \mathit{acc}\ [\,] = \mathit{acc}\\
&\qquad\qquad \mathit{rev}\ \mathit{op}\ \mathit{acc}\ (x : \mathit{xs}) = \mathit{rev}\ \mathit{op}\ (\mathit{acc}\ `\mathit{op}`\ x)\ \mathit{xs}
\end{aligned}
$$

which is the same as:

$$reverse\ xs = foldl\ revOp\ [\,]\ xs$$

If all of this seems like magic, well, you are starting to see the beauty of functional programming!

3.6 Errors

In the last section we talked about the idea of "returning an error" when the argument to *foldr1* is the empty list. As you might imagine, there are other situations where an error result is also warranted.

There are many ways to deal with such situations, depending on the application, but sometimes we wish to literally stop the program, signalling to the user that some kind of an *error* has occurred. In Haskell this is done with the Standard Prelude function $error :: String \rightarrow a$. Note that *error* is polymorphic, meaning that it can be used with any data type. The value of the expression *error s* is *bottom*, the completely undefined, or "bottom" value. As an example of its use, here is the definition of *foldr1* from the Standard Prelude:

```
foldr1 :: (a -> a -> a) -> [a] -> a
foldr1 f [x] = x
foldr1 f (x : xs) = f x (foldr1 f xs)
foldr1 f [] = error "Prelude.foldr1: empty list"
```

Thus if the anomalous situation arises, the program will terminate immediately, and the string `"Prelude.foldr1: empty list"` will be printed.

Exercise 3.1 Rewrite the equation for the area of a polygon given in Section **??** in a higher-order, non-recursive way, using operators such as *map* and *fold*, etc.

Exercise 3.2 What is the principal type of each of the following expressions:

```
map map
map foldl
```

Exercise 3.3 Rewrite the definition of *length* non-recursively.

Exercise 3.4 Define a function that behaves as each of the following:

1. Doubles each number in a list. For example:

   $$\mathit{doubleEach}\ [1,2,3] \Longrightarrow [2,4,6]$$

2. Pairs each element in a list with that number and one plus that number. For example:

   $$\mathit{pairAndOne}\ [1,2,3] \Longrightarrow [(1,2),(2,3),(3,4)]$$

3. Adds together each pair of numbers in a list. For example:

   $$\mathit{addEachPair}\ [(1,2),(3,4),(5,6)] \Longrightarrow [3,7,11]$$

In this exercise and the two that follow, give both recursive and (if possible) non-recursive definitions, and be sure to include type signatures.

Exercise 3.5 Define a function *maxList* that computes the maximum element of a list. Define *minList* analogously.

Exercise 3.6 Define a function that adds "pointwise" the elements of a list of pairs. For example:

$$\mathit{addPairsPointwise}\ [(1,2),(3,4),(5,6)] \Longrightarrow (9,12)$$

Exercise 3.7 Freddie the Frog wants to communicate privately with his girlfriend Francine by *encrypting* messages sent to her. Frog brains are not that large, so they agree on this simple strategy: each character in the text shall be converted to the character "one greater" than it, based on the representation described below (with wrap-around from 255 to 0). Define functions *encrypt* and *decrypt* that will allow Freddie and Francine to communicate using this strategy.

Hint: Characters are often represented inside a computer as some kind of an integer; in the case of Haskell, a 16-bit unicode representation is used. For this exercise, you will want to use two Haskell functions, *toEnum* and *fromEnum*. The first will convert an integer into a character, the second will convert a character into an integer.

Exercise 3.8 Suppose you are given a non-negative integer *amt* representing a sum of money, and a list of coin denominations $[v1, v2, ..., vn]$, each being a positive integer. Your job is to make change for *amt* using the coins in the coin supply. Define a function *makeChange* to solve this problem. For example, your function may behave like this:

$makeChange\ 99\ [5,1] \Rightarrow [19,4]$

where 99 is the amount and $[5,1]$ represents the types of coins (say, nickels and pennies in US currency) that we have. The answer $[19,4]$ means that we can make the exact change with 19 5-unit coins and 4 single-unit coins; this is the best (in terms of the total number of coins) possible solution.

To make things slightly easier, you may assume that the list representing the coin denominations is given in descending order, and that the single-unit coin is always one of the coin types.

[Need to add some musical exercises.]

Chapter 4

More About Higher-Order Functions

You have now seen several examples where functions are passed as arguments to other functions, such as with *fold* and *map*. In this chapter I will show several examples where functions are also returned as values. This will lead to several techniques for improving definitions that we have already written, techniques that we will use often in the remainder of the text.

4.1 Currying

The first improvement relates to the notation we have used to write function applications, such as *simple x y z*. Although I have noted the similarity of this to the mathematical notation $simple(x, y, z)$, in fact there is an important difference, namely that *simple x y z* is actually equivalent to (((*simple x*) *y*) *z*). In other words, function application is *left associative*, taking one argument at a time.

Let's look at the expression (((*simple x*) *y*) *z*) a bit closer: there is an application of *simple* to x, the result of which is applied to y; so (*simple x*) must be a function! The result of this application, ((*simple x*) *y*), is then applied to z, so ((*simple x*) *y*) must also be a function!

Since each of these intermediate applications yields a function, it seems perfectly reasonable to define a function such as:

multSumByFive = *simple* 5

What is *simple* 5? From the above argument we know that it must be a function. And from the definition of *simple* in Section 1.1 we might guess that

this function takes two arguments, and returns 5 times their sum. Indeed, we can *calculate* this result as follows:

$$
\begin{aligned}
&\mathit{multSumByFive}\ a\ b \\
&\Rightarrow (\mathit{simple}\ 5)\ a\ b \\
&\Rightarrow \mathit{simple}\ 5\ a\ b \\
&\Rightarrow 5 * (a + b)
\end{aligned}
$$

The intermediate step with parentheses is included just for clarity. This method of applying functions to one argument at a time, yielding intermediate functions along the way, is called *currying*, after the logician Haskell B. Curry who popularized the idea.[1] It is helpful to look at the types of the intermediate functions as arguments are applied:

$$
\begin{aligned}
&\mathit{simple} :: \mathit{Float} \rightarrow \mathit{Float} \rightarrow \mathit{Float} \rightarrow \mathit{Float} \\
&\mathit{simple}\ 5 :: \mathit{Float} \rightarrow \mathit{Float} \rightarrow \mathit{Float} \\
&\mathit{simple}\ 5\ a :: \mathit{Float} \rightarrow \mathit{Float} \\
&\mathit{simple}\ 5\ a\ b :: \mathit{Float}
\end{aligned}
$$

How can we use currying to improve any of our previous examples? One place is in these definitions of *line* and *maxPitch*:

$$
\begin{aligned}
&\mathit{line} :: [\mathit{Music}] \rightarrow \mathit{Music} \\
&\mathit{line}\ \mathit{ms} = \mathit{fold}\ (:+:)\ (\mathit{Primitive}\ (\mathit{Rest}\ 0))\ \mathit{ms}
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{maxPitch} :: [\mathit{Pitch}] \rightarrow \mathit{Pitch} \\
&\mathit{maxPitch}\ \mathit{ps} = \mathit{fold}\ (!!!)\ 0\ \mathit{ps}
\end{aligned}
$$

which can be simplified to:

$$
\begin{aligned}
&\mathit{line} :: [\mathit{Music}] \rightarrow \mathit{Music} \\
&\mathit{line} = \mathit{fold}\ (:+:)\ (\mathit{Primitive}\ (\mathit{Rest}\ 0))
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{maxPitch} :: [\mathit{Pitch}] \rightarrow \mathit{Pitch} \\
&\mathit{maxPitch} = \mathit{fold}\ (!!!)\ 0
\end{aligned}
$$

Similarly, this defintion of *listSum*:

$$
\begin{aligned}
&\mathit{listSum} :: [\mathit{Float}] \rightarrow \mathit{Float} \\
&\mathit{listSum}\ \mathit{xs} = \mathit{foldl}\ (+)\ 0\ \mathit{xs}
\end{aligned}
$$

[1] It was actually Schönfinkel who first called attention to this idea [**?**], but the word "schönfinkelling" is rather a mouthful!

can be simplified to:

$$\begin{aligned} &\mathit{listSum} :: [\mathit{Float}] \to \mathit{Float} \\ &\mathit{listSum} = \mathit{foldl}\ (+)\ 0 \end{aligned}$$

We will refer to this kind of simplification as "currying simplification" or just "currying," even though it actually has a more technical name, "eta contraction."

> **Details:** Some care should be taken when using this simplification idea. In particular, note that an equation such as $f\ x = g\ x\ y\ x$ cannot be simplified to $f = g\ x\ y$, since then the x would become undefined!

Here is a more interesting example, in which I will use currying simplification three times. Recall from Section 3.5 the definition of *reverse* using *foldl*:

$$\begin{aligned} &\mathit{reverse}\ \mathit{xs} = \mathit{foldl}\ \mathit{revOp}\ [\,]\ \mathit{xs} \\ &\quad \textbf{where}\ \mathit{revOp}\ \mathit{acc}\ x = x : \mathit{acc} \end{aligned}$$

Using the polymorphic function *flip* which is defined in the Standard Prelude as:

$$\begin{aligned} &\mathit{flip} :: (a \to b \to c) \to (b \to a \to c) \\ &\mathit{flip}\ f\ x\ y = f\ y\ x \end{aligned}$$

it should be clear that *revOp* can be rewritten as:

$$\mathit{revOp}\ \mathit{acc}\ x = \mathit{flip}\ (:)\ \mathit{acc}\ x$$

But now currying simplification can be used twice to reveal that:

$$\mathit{revOp} = \mathit{flip}\ (:)$$

This, along with a third use of currying, allows us to rewrite the definition of *reverse* simply as:

$$\mathit{reverse} = \mathit{foldl}\ (\mathit{flip}\ (:))\ [\,]$$

This is in fact the way *reverse* is defined in the Standard Prelude.

Exercise 4.1 Show that $\mathit{flip}\ (\mathit{flip}\ f)$ is the same as f.

Exercise 4.2 What is the type of *ys* in:

$xs = [1, 2, 3] :: [Float]$
$ys = map\ (+)\ xs$

Exercise 4.3 Define a function *applyEach* that, given a list of functions, applies each to some given value. For example:

$applyEach\ [simple\ 2\ 2, (+3)]\ 5 \Longrightarrow [14, 8]$

where *simple* is as defined in Section 1.1.

Exercise 4.4 Define a function *applyAll* that, given a list of functions $[f1, f2, ..., fn]$ and a value v, returns the result $f1\ (f2\ (...(fn\ v)...))$. For example:

$applyAll\ [simple\ 2\ 2, (+3)]\ 5 \Longrightarrow 20$

Exercise 4.5 Recall the discussion about the efficiency of $(+\!\!+)$ and *concat* in Chapter 3. Which of the following functions is more efficient, and why?

$appendr, appendl :: [[a]] \rightarrow [a]$
$appendr = foldr\ (flip\ (+\!\!+))\ [\,]$
$appendl = foldl\ (flip\ (+\!\!+))\ [\,]$

4.2 Sections

With a bit more syntax, we can also curry applications of infix operators such as $(+)$. This syntax is called a *section*, and the idea is that, in an expression such as $(x + y)$, you can omit either the x or the y, and the result (with the parentheses still intact) is a function of that missing argument. If *both* variables are omitted, it is a function of *two* arguments. In other words, the expressions $(x+)$, $(+y)$ and $(+)$ are equivalent, respectively, to the functions:

$f1\ y = x + y$
$f2\ x = x + y$
$f3\ x\ y = x + y$

For example, suppose that we need to determine whether each number in a list is positive. Instead of writing:

$posInts :: [Integer] \rightarrow [Bool]$
$posInts\ xs = map\ test\ xs$
$\qquad\qquad$ **where** $test\ x = x > 0$

we can simply write:

$$posInts :: [Integer] \rightarrow [Bool]$$
$$posInts\ xs = map\ (>0)\ xs$$

which can be further simplified using currying:

$$posInts :: [Integer] \rightarrow [Bool]$$
$$posInts = map\ (>0)$$

This is an extremely concise definition.

As you gain experience with higher-order functions you will not only be able to start writing definitions such as this directly, but you will also start *thinking* in "higher-order" terms. We will see many examples of this kind of reasoning throughout the text.

Exercise 4.6 Define a function *twice* that, given a function f, returns a function that applies f twice to its argument. For example:

$$(twice\ (+1))\ 2 \Rightarrow 4$$

What is the principal type of *twice*? Describe what *twice twice* does, and give an example of its use. How about *twice twice twice* and *twice* (*twice twice*)?

Exercise 4.7 Generalize *twice* defined in the previous exercise by defining a function *power* that takes a function f and an integer n, and returns a function that applies the function f to its argument n times. For example:

$$power\ (+2)\ 5\ 1 \Longrightarrow 11$$

Use *power* to define something (anything!) useful.

4.3 Anonymous Functions

The final way to define a function in Haskell is in some sense the most fundamental: it is called an *anonymous function*, or *lambda expressions* (since the concept is drawn directly from Church's lambda calculus [**?**]). The idea is that functions are values, just like numbers and characters and strings, and therefore there should be a way to create them without having to give them a name. As a simple example, an anonymous function that increments its numeric argument by one can be written $\lambda x \rightarrow x + 1$. Anonymous functions are most useful in situations where you don't wish to name them, which is why they are called "anonymous."

Details: The typesetting that we use in this book prints an actual Greek lambda character, but in writing $\lambda x \rightarrow x + 1$ in your programs you will have to write "`\x -> x+1`" instead.

As another example, to add one and then divide by two every element of a list, we could write:

$$map\ (\lambda x \rightarrow (x + 1)\ /\ 2)\ xs$$

An even better example is an anonymous function that pattern-matches its argument, as in:

$$map\ (\lambda(a, b) \rightarrow a + b)\ xs$$

Details: Anonymous functions can only perform one match against an argument. That is, you cannot stack together several anonymous functions to define one function, as you can with equations.

Anonymous functions are considered most fundamental because definitions such as that for *simple* given in Chapter 1:

$$simple\ x\ y\ z = x * (y + z)$$

can be written instead as:

$$simple = \lambda x\ y\ z \rightarrow x * (y + z)$$

Details: $\lambda x\ y\ z \rightarrow exp$ is shorthand for $\lambda x \rightarrow \lambda y \rightarrow \lambda z \rightarrow exp$.

We can also use anonymous functions to explain precisely the behavior of sections. In particular, note that:

$$(x+) \Rightarrow \lambda y \rightarrow x + y$$
$$(+y) \Rightarrow \lambda x \rightarrow x + y$$
$$(+) \Rightarrow \lambda x\ y \rightarrow x + y$$

Exercise 4.8 Suppose we define a function *fix* as:

$$fix\ f = f\ (fix\ f)$$

What is the principal type of *fix*? (This is tricky!) Suppose further that we have a recursive function:

$$remainder :: Integer \rightarrow Integer \rightarrow Integer$$
$$remainder\ a\ b = \textbf{if}\ a < b\ \textbf{then}\ a$$
$$\textbf{else}\ remainder\ (a - b)\ b$$

Rewrite this function using *fix* so that it is not recursive. (Also tricky!) Do you think that this process can be applied to *any* recursive function?

4.4 Function Composition

Figure 4.1: Gluing Two Functions Together

An example of polymorphism that has nothing to do with data structures arises from the desire to take two functions f and g and "glue them together," yielding another function h that first applies g to its argument, and then applies f to that result. This is called function *composition*, and Haskell pre-defines a simple infix operator $(\circ)$ to achieve it, as follows:

$$(\circ) :: (b \to c) \to (a \to b) \to a \to c$$
$$(f \circ g)\ x = f\ (g\ x)$$

> **Details:** The symbol for function composition is typeset in this book as $\circ$, which is the proper mathematical convention. When writing your programs, however, you will have to use a "period" , as in "`f . g`".

Note the type of the operator $(\circ)$; it is completely polymorphic. Note also that the result of the first function to be applied—some type b—must be the same as the type of the argument to the second function to be applied. Pictorially, if you think of a function as a black box that takes input at one end and returns some output at the other, function composition is like connecting two boxes together, end to end, as shown in Figure 4.1.

The ability to compose functions using $(\circ)$ is extremely useful. For example, consider this function to compute the sum of the areas of circles with various radii:

$$totalCircleArea :: [Float] \to Float$$
$$totalCircleArea\ radii = listSum\ (map\ circleArea\ radii)$$

We can first add parentheses to emphasize the application of interest:

$$totalCircleArea :: [Float] \to Float$$
$$totalCircleArea\ radii = listSum\ ((map\ circleArea)\ radii)$$

then rewrite as a function composition:

$$totalCircleArea :: [Float] \to Float$$
$$totalCircleArea\ radii = (listSum \circ (map\ circleArea))\ radii$$

and finally use currying to simplify:

$totalCircleArea :: [Float] \rightarrow Float$
$totalCircleArea = listSum \circ map\ circleArea$

Similarly, this definition:

$totalSquareArea :: [Float] \rightarrow Float$
$totalSquareArea\ sides = listSum\ (map\ squareArea\ sides)$

can be rewritten as:

$totalSquareArea :: [Float] \rightarrow Float$
$totalSquareArea = listSum \circ map\ squareArea$

But let's also create additional compositions. A function that determines whether all of the elements in a list are greater than zero, and one that determines if at least one is greater than zero, can be written:

$allOverZero, oneOverZero :: [Integer] \rightarrow Bool$
$allOverZero = and \circ posInts$
$oneOverZero = or \circ posInts$

Note that the auxiliary function *posInts* is simple enough that we could incorporate its definition directly, as in:

$allOverZero, oneOverZero :: [Integer] \rightarrow Bool$
$allOverZero = and \circ map\ (>0)$
$oneOverZero = or \circ map\ (>0)$

In the remainder of this text I will not refrain from writing definitions such as this directly, using a small set of rich polymorphic functions such as *fold* and *map*, plus a few others drawn from the Prelude and Standard Libraries.

Exercise 4.9 Rewrite this example:

$map\ (\lambda x \rightarrow (x + 1)\ /\ 2)\ xs$

using a composition of sections.

Exercise 4.10 Consider the expression:

$map\ f\ (map\ g\ xs)$

Rewrite this using function composition and a single call to *map*. Then rewrite the earlier example:

$map\ (\lambda x \rightarrow (x + 1)\ /\ 2)\ xs$

as a "map of a map."

Exercise 4.11 Go back to any exercises prior to this chapter, and simplify your solutions using ideas learned here.

Exercise 4.12 Using higher-order functions that we have now defined, fill in the two missing functions, *f1* and *f2*, in the evaluation below so that it is valid:

$$f1\ (f2\ (*)\ [1,2,3,4])\ 5 \Rightarrow [5,10,15,20]$$

Chapter 5

More Music

```
module Haskore.MoreMusic where
import Haskore.Music
import Ratio
import Ix
```

In this chapter we will explore a numer of simple musical ideas, and contribute to a growing collection of Haskell functions for expressing those ideas.

5.1 Delay and Repeat

Suppose that we wish to describe a melody m accompanied by an identical voice a perfect 5th higher. In Haskore we can simply write $m :=: transpose\ 7\ m$. Similarly, a canon-like structure involving m can be expressed as $m :=: delay\ d\ m$, where:

```
delay :: Dur → Music a → Music a
delay d m = rest d :+: m
```

More interestingly, Haskell's non-strict semantics also allows us to define *infinite* musical values. For example, a musical value may be repeated *ad nauseum* using this simple function:

```
repeatM :: Music a → Music a
repeatM m = m :+: repeatM m
```

Thus, for example, an infinite ostinato can be expressed in this way, and then used in different contexts that automatically extract only the portion that is actually needed. We will see more examples of this shortly.

Figure 5.1: Nested Polyrhythms

5.2 Inversion and Retrograde

The notions of inversion, retrograde, retrograde inversion, etc. as used in 12-tone theory are also easily captured in Haskore. [insert explanation of these concepts]

First let's define a transformation from a line created by *line* to a list:

```
lineToList :: Music a → [Music a]
lineToList n@(Primitive (Rest 0)) = []
lineToList (n :+: ns) = n : lineToList ns
lineToList _ = error "lineToList: argument not created by line"
```

Using this function it is then straightforward to define *invert*, from which the other functions are easily defined via composition:

```
retro, invert, retroInvert, invertRetro :: Music Pitch → Music Pitch
invert m = line (map inv l)
  where l@(Primitive (Note _ r) : _) = lineToList m
        inv (Primitive (Note d p)) =
                        note d (pitch (2 * absPitch r − absPitch p))
        inv (Primitive (Rest d)) = rest d
retro = line ∘ reverse ∘ lineToList
retroInvert = retro ∘ invert
invertRetro = invert ∘ retro
```

Exercise 5.1 *Show that retro ∘ retro, invert ∘ invert, and retroInvert ∘ invertRetro are the identity on values created by line.*

5.3 Polyrhythms

For some rhythmical ideas, first note that if m is a line of three eighth notes, then *tempo* $(3 \% 2)$ m is a *triplet* of eighth notes. In fact *tempo* can be used to create quite complex rhythmical patterns. For example, consider the "nested polyrhythms" shown in Figure 5.1. They can be expressed naturally in Haskore as follows (note the use of the *where* clause in *pr2* to capture recurring phrases):

```
pr1, pr2 :: Pitch → Music Pitch
pr1 p = tempo (5 % 6)
            (tempo (4 % 3) (mkLn 1 p qn:+:
                                tempo (3 % 2) (mkLn 3 p en:+:
                                                   mkLn 2 p sn:+:
                                                   mkLn 1 p qn):+:
                                mkLn 1 p qn):+:
             tempo (3 % 2) (mkLn 6 p en))

pr2 p = tempo (7 % 6) (m1:+:
                       tempo (5 % 4) (mkLn 5 p en):+:
                       m1:+:
                       mkLn 2 p en)
        where m1 = tempo (5 % 4) (tempo (3 % 2) m2 :+: m2)
              m2 = mkLn 3 p en

mkLn n p d = line (take n (repeat (note d p)))
```

Details: *take n lst* is the first n elements of the list *lst*. For example, *take* $3\ [C, Cs, Df, D, Ds] \Longrightarrow [C, Cs, Df]$. *repeat x* is the infinite list of the same value x. For example, *take* 3 (*repeat* 42) $\Longrightarrow [42, 42, 42]$.

To play polyrhythms *pr1* and *pr2* in parallel using middle C and middle G, respectively, we do the following:

```
pr12 :: Music Pitch
pr12 = pr1 (C, 4) :=: pr2 (G, 4)
```

5.4 Symbolic Meter Changes

We can implement the notion of "symbolic meter changes" of the form "old-note = newnote" (quarter note = dotted eighth, for example) by defining an infix function:

```
(=:=) :: Dur → Dur → Music a → Music a
old =:= new = tempo (new / old)
```

Of course, using the new function is not much longer than using *Tempo* directly, but it may have nemonic value.

5.5 Computing Duration

It is often desirable to compute the *duration*, in whole notes, of a musical value; we can do so as follows:

```
dur :: Music a → Dur
dur (Primitive (Note d _)) = d
dur (Primitive (Rest d)) = d
dur (m1 :+: m2) = dur m1 + dur m2
dur (m1 :=: m2) = dur m1 `max` dur m2
dur (Modify (Tempo r) m) = dur m / r
dur (Modify _ m) = dur m
```

5.6 Super-retrograde

Using *dur* we can define a function *revM* that reverses any *Music* value (and is thus considerably more useful than *retro* defined earlier). Note the tricky treatment of (:=:).

```
revM :: Music a → Music a
revM n@(Primitive _) = n
revM (Modify c m) = Modify c (revM m)
revM (m1 :+: m2) = revM m2 :+: revM m1
revM (m1 :=: m2) =
  let d1 = dur m1
      d2 = dur m2
  in if d1 > d2 then revM m1 :=: (rest (d1 - d2) :+: revM m2)
                else (rest (d2 - d1) :+: revM m1) :=: revM m2
```

5.7 Truncating Parallel Composition

Note that the duration of *m1* :=: *m2* is the maximum of the durations of *m1* and *m2* (and thus if one is infinite, so is the result). Sometimes we would rather have the result be of duration equal to the shorter of the two. This is not as easy as it sounds, since it may require interrupting the longer one in the middle of a note (or notes).

We will define a "truncating parallel composition" operator (/=:), but first we will define an auxiliary function *cut* such that *cut d m* is the musical value *m* "cut short" to have at most duration *d*:

```
cut :: Dur → Music a → Music a
cut newDur m | newDur ⩽ 0 = rest 0
cut newDur (Primitive (Note oldDur x)) = note (min oldDur newDur) x
cut newDur (Primitive (Rest oldDur)) = rest (min oldDur newDur)
cut newDur (m1 :=: m2) = cut newDur m1 :=: cut newDur m2
cut newDur (m1 :+: m2) = let m1' = cut newDur m1
                             m2' = cut (newDur − dur m1') m2
                         in m1' :+: m2'
cut newDur (Modify (Tempo r) m) = tempo r (cut (newDur ∗ r) m)
cut newDur (Modify c m) = Modify c (cut newDur m)
```

Note that *cut* is equipped to handle a *Music* value of infinite length.

With *cut*, the definition of (/=:) is now straightforward:

```
(/=:) :: Music a → Music a → Music a
m1 /=: m2 = cut (min (dur m1) (dur m2)) (m1 :=: m2)
```

Unfortunately, whereas *cut* can handle infinite-duration music values, (/=:) cannot.

Exercise 5.2 *Define a version of* (/=:) *that shortens correctly when either or both of its arguments are infinite in duration.*

5.8 Trills

A *trill* is an ornament that alternates rapidly between two (usually adjacent) pitches. We will define two versions of a trill function, both of which take the starting note and an interval for the trill note as arguments (the interval is usually one or two, but can actually be anything). One version will additionally have an argument that specifies how long each trill note should be, whereas the other will have an argument that specifies how many trills should occur. In both cases the total duration will be the same as the duration of the original note.

Here is the first trill function:

```
trill :: Int → Dur → Music Pitch → Music Pitch
trill i sDur (Primitive (Note tDur p)) =
    if sDur ⩾ tDur then note tDur p
    else note sDur p
         :+: trill (negate i) sDur (note (tDur − sDur) (trans i p))
trill i d (Modify (Tempo r) m) = tempo r (trill i (d ∗ r) m)
```

```
trill i d (Modify c m) = Modify c (trill i d m)
trill _ _ _ = error "trill: input must be a single note."
```

It is simple to define a version of this function that starts on the trill note rather than the start note:

```
trill' :: Int → Dur → Music Pitch → Music Pitch
trill' i sDur m = trill (negate i) sDur (transpose i m)
```

The second way to define a trill is in terms of the number of subdivided notes to be included in the trill. We can use the first trill function to define this new one:

```
trilln :: Int → Int → Music Pitch → Music Pitch
trilln i nTimes m = trill i (dur m / fromIntegral nTimes) m
```

This, too, can be made to start on the other note.

```
trilln' :: Int → Int → Music Pitch → Music Pitch
trilln' i nTimes m = trilln (negate i) nTimes (transpose i m)
```

Finally, a *roll* can be implemented as a trill whose interval is zero. This feature is particularly useful for percussion.

```
roll :: Dur → Music Pitch → Music Pitch
rolln :: Int → Music Pitch → Music Pitch

roll dur m = trill 0 dur m
rolln nTimes m = trilln 0 nTimes m
```

5.9 Percussion

Percussion is a difficult notion to represent in the abstract. On one hand, a percussion instrument is just another instrument, so why should it be treated differently? On the other hand, even common practice notation treats it specially, even though it has much in common with non-percussive notation. The midi standard is equally ambiguous about the treatment of percussion: on one hand, percussion sounds are chosen by specifying an octave and pitch, just like any other instrument; on the other hand these pitches have no tonal meaning whatsoever: they are just a convenient way to select from a large number of percussion sounds. Indeed, part of the General MIDI Standard is a set of names for commonly used percussion sounds.

```
data PercussionSound =
        AcousticBassDrum      -- MIDI Key 35
      | BassDrum1      -- MIDI Key 36
      | SideStick      -- ...
      | AcousticSnare | HandClap | ElectricSnare | LowFloorTom
      | ClosedHiHat | HighFloorTom | PedalHiHat | LowTom
      | OpenHiHat | LowMidTom | HiMidTom | CrashCymbal1
      | HighTom | RideCymbal1 | ChineseCymbal | RideBell
      | Tambourine | SplashCymbal | Cowbell | CrashCymbal2
      | Vibraslap | RideCymbal2 | HiBongo | LowBongo
      | MuteHiConga | OpenHiConga | LowConga | HighTimbale
      | LowTimbale | HighAgogo | LowAgogo | Cabasa
      | Maracas | ShortWhistle | LongWhistle | ShortGuiro
      | LongGuiro | Claves | HiWoodBlock | LowWoodBlock
      | MuteCuica | OpenCuica | MuteTriangle
      | OpenTriangle      -- MIDI Key 82
  deriving (Show, Eq, Ord, Ix, Enum)
```

Figure 5.2: General MIDI Percussion Names

Since MIDI is such a popular platform, we can at least define some handy functions for using the General MIDI Standard. We start by defining the data type shown in Figure 5.2, which borrows its constructor names from the General MIDI standard. The comments reflecting the "MIDI Key" numbers will be explained later, but basically a MIDI Key is the equivalent of an absolute pitch in Haskore terminology. So all we need is a way to convert these percussion sound names into a *Music* value; i.e. a *Note*:

```
perc :: PercussionSound → Dur → Music Pitch
perc ps dur = note dur (pitch (fromEnum ps + 35))
```

> **Details:** *fromEnum* is a method in the *Enum* class, which is all about enumerations. A data type that is a member of this class can be *enumerated*—i.e. the elements of the data type can be listed in order. *fromEnum* maps each value to its index in this enumeration. Thus *fromEnum AcousticBassDrum* is 0, *fromEnum BassDrum1* is 1, and so on.

For example, here are eight bars of a simple rock or "funk groove" that uses *perc* and *roll*:

```
funkGroove
  = let p1 = perc LowTom qn
```

```
    p2 = perc AcousticSnare en
in tempo 3 (instrument Percussion (cut 8 (repeatM
      ((p1 :+: qnr :+: p2 :+: qnr :+: p2:+:
        p1 :+: p1 :+: qnr :+: p2 :+: enr)
       :=: roll en (perc ClosedHiHat 2))
  )))
```

Exercise 5.3 *Find a simple piece of music written by your favorite composer, and transcribe it into Haskore. In doing so, look for repeating patterns, transposed phrases, etc. and reflect this in your code, thus revealing deeper structural aspects of the music than that found in common practice notation.*

Chapter 6

Interpretation and Performance

```
module Haskore.Performance
       where

import Haskore.Music
import Haskore.MoreMusic

instance Show (a → b) where
    showsPrec p f = showString "<<function>>"
```

6.1 Abstract Performance

So far, our presentation of musical values in Haskell has been entirely structural, i.e. *syntactic*. But what do these musical values actually mean, i.e. what is their *semantics*, or *interpretation*? The formal process of giving a semantic interpretation to syntactic constructs is very common in computer science, especially in programming language theory. But it is obviously also common in music: the interpretation of music is the very essence of musical performance. However, in conventional music this process is usually informal, appealing to aesthetic judgments and values. What we would like to do is make the process formal in Haskore—but still flexible, so that more than one interpretation is possible, just as in music.

To begin, we need to say exactly what an abstract *performance* is. Our approach is to consider a performance to be a time-ordered sequence of musical *events*, where each event captures the playing of one individual

note. In Haskellese:

$$\textbf{type } \mathit{Performance} = [\mathit{Event}]$$

$$\begin{aligned}\textbf{data } \mathit{Event} = \mathit{Event}\{\, &\mathit{eTime} :: \mathit{Time}, \mathit{eInst} :: \mathit{InstrumentName}, \mathit{ePitch} :: \mathit{AbsPitch},\\ &\mathit{eDur} :: \mathit{DurT}, \mathit{eVol} :: \mathit{Volume}, \mathit{pFields} :: [\mathit{Float}]\}\\ &\textbf{deriving } (\mathit{Eq}, \mathit{Ord}, \mathit{Show})\end{aligned}$$

$$\textbf{type } \mathit{Time} = \mathit{Rational}$$
$$\textbf{type } \mathit{DurT} = \mathit{Rational}$$
$$\textbf{type } \mathit{Volume} = \mathit{Int}$$

An event $\mathit{Event}\{\, \mathit{eTime} = s, \mathit{eInst} = i, \mathit{ePitch} = p, \mathit{eDur} = d, \mathit{eVol} = v\}$ captures the fact that at start time s, instrument i sounds pitch p with volume v for a duration d (where now duration is measured in seconds, rather than beats). (The *pField* of an event is for special instruments that require extra parameters, and will not be discussed much further in this chapter.)

An abstract performance is the lowest of our music representations not yet committed to MIDI, csound, or some other low-level computer music representation. In a later chapter we will discuss how to map a performance into MIDI.

Details: The data declaration for Event uses Haskell's *field label* syntax, also called *record* syntax, and is equivalent to:

$$\begin{aligned}&\textbf{data } \mathit{Event} = \mathit{Event}\ \mathit{Time}\ \mathit{InstrumentName}\ \mathit{AbsPitch}\ \mathit{DurT}\ \mathit{Volume}\ [\mathit{Float}]\\ &\quad\textbf{deriving } (\mathit{Eq}, \mathit{Ord}, \mathit{Show})\end{aligned}$$

except that the former also defines "field labels" eTime, eInst, ePitch, eDur, eVol, and $\mathit{pFields}$, which can be used both to create and select from Event values. For example, this equation:

$$e = \mathit{Event}\ 0\ \mathit{Cello}\ 27\ (1 \,\%\, 4)\ 50\ [\,]$$

is equivalent to:

$$\begin{aligned}e = \mathit{Event}\{\, &\mathit{eTime} = 0, \mathit{eInst} = \mathit{Cello}, \mathit{ePitch} = 27,\\ &\mathit{eDur} = 1 \,\%\, 4, \mathit{eVol} = 50, \mathit{pFields} = [\,]\}\end{aligned}$$

The latter is more descriptive, however, and the order of the fields does not matter.

Field labels can be used to select fields from an Event value; for example, $\mathit{eInst}\ e \Rightarrow \mathit{Cello}$, $\mathit{eDur}\ e \Rightarrow 1 \,\%\, 4$, and so on. They can also be used to selectively *update* fields of an existing Event value. For example:

$$e\{\, eInst = Flute \,\} \Rightarrow Event\ 0\ Flute\ 27\ (1\ \%\ 4)\ 50\ [\,]$$

Finally, they can be used selectively in pattern matching:

$$f\ (Event\{\, eDur = d, ePitch = p \,\}) = ...d\ ...\ p...$$

Field labels do not change the basic nature of a data type; they are simply a convenient syntax for referring to the components of a data type by name rather than by position.

To generate a complete performance of, i.e. give an interpretation to, a musical value, we must know the time to begin the performance, and the proper instrument, volume, key and tempo. In addition, to give flexibility to our interpretations, we must also know what *player* to use; that is, we need a mapping from the *PlayerName*s in a *Music* value to the actual players to be used.[1] We capture these ideas in Haskell as a "context" and "player map":

```
data Context a = Context{ cTime :: Time, cPlayer :: Player a,
                          cInst :: InstrumentName, cDur :: DurT,
                          cKey :: Key, cVol :: Volume }
    deriving Show
type PMap a = PlayerName → Player a
type Key = AbsPitch
```

Finally, we are ready to give an interpretation to a piece of music, which we do by defining a function *perform*, which is conceptually perhaps the most important function defined in this book:

```
perform :: PMap a → Context a → Music a → Performance
perform pmap
  c@Context{ cTime = t, cPlayer = pl, cDur = dt, cKey = k }m =
  case m of
    Primitive (Note d p) → playNote pl c d p
    Primitive (Rest d) → [ ]
    m1 :+: m2 → perform pmap c m1 ++
                perform pmap (c{ cTime = t + dur m1 * dt }) m2
    m1 :=: m2 → merge (perform pmap c m1) (perform pmap c m2)
    Modify (Tempo r) m → perform pmap (c{ cDur = dt / r }) m
```

[1] We don't need a mapping from *InstrumentNames* to instruments, since this is handled in the translation from a performance into, say, MIDI.

```
perform :: PMap a → Context a → Music a → Performance
perform pmap c m = fst (perf pmap c m)

perf :: PMap a → Context a → Music a → (Performance, DurT)
perf pmap
  c@Context{ cTime = t, cPlayer = pl, cDur = dt, cKey = k }m =
  case m of
    Primitive (Note d p) → (playNote pl c d p, d * dt)
    Primitive (Rest d) → ([ ], d * dt)
    m1 :+: m2 → let (pf1, d1) = perf pmap c m1
                    (pf2, d2) = perf pmap (c{ cTime = t + d1 }) m2
                in (pf1 ++ pf2, d1 + d2)
    m1 :=: m2 → let (pf1, d1) = perf pmap c m1
                    (pf2, d2) = perf pmap c m2
                in (merge pf1 pf2, max d1 d2)
    Modify (Tempo r) m → perf pmap (c{ cDur = dt / r }) m
    Modify (Transpose p) m → perf pmap (c{ cKey = k + p }) m
    Modify (Instrument i) m → perf pmap (c{ cInst = i }) m
    Modify (Player pn) m → perf pmap (c{ cPlayer = pmap pn }) m
    Modify (Phrase pas) m → interpPhrase pl pmap c pas m
```

Figure 6.1: The "real" *perform* function.

```
Modify (Transpose p) m → perform pmap (c{ cKey = k + p }) m
Modify (Instrument i) m → perform pmap (c{ cInst = i }) m
Modify (Player pn) m → perform pmap (c{ cPlayer = pmap pn }) m
Modify (Phrase pa) m → interpPhrase pl pmap c pa m
```

Some things to note about *perform*:

1. The *Context* is the running "state" of the performance, and gets updated in several different ways. For example, the interpretation of the *Tempo* constructor involves scaling *dt* appropriately and updating the *DurT* field of the context.

2. The interpretation of notes and phrases is player dependent. Ultimately a single note is played by the *playNote* function, which takes the player as an argument. Similarly, phrase interpretation is also player dependent, reflected in the use of *interpPhrase*. Precisely how these two functions work is described in Section 6.2.

3. The *DurT* component of the context is the duration, in seconds, of one whole note. To make it easier to compute, we can define a "metronome" function that, given a standard metronome marking (in beats per minute) and the note type associated with one beat (quarter note, eighth note, etc.) generates the duration of one whole note:

   $$\begin{aligned} &metro :: Int \to Dur \to DurT \\ &metro\ setting\ dur = 60\ /\ (fromIntegral\ setting * dur) \end{aligned}$$

   Thus, for example, *metro* 96 *qn* creates a tempo of 96 quarter notes per minute.

4. In the treatment of (:+:), note that the sub-sequences are appended together, with the start time of the second argument delayed by the duration of the first. The function *dur* (defined in Section **??**) is used to compute this duration. However, this results in a quadratic time complexity for *perform*. A more efficient solution is to have *perform* compute the duration directly, returning it as part of its result. This version of *perform* is shown in Figure 6.1.

5. The sub-sequences derived from the arguments to (:=:) are merged into a time-ordered stream. The definition of *merge* is given below.

$$\begin{aligned} &merge :: Performance \to Performance \to Performance \\ \\ &merge\ a@(e1 : es1)\ b@(e2 : es2) = \\ &\quad \textbf{if}\ e1 < e2\ \textbf{then}\ e1 : merge\ es1\ b \\ &\qquad\qquad \textbf{else}\ e2 : merge\ a\ es2 \\ &merge\ [\,]\ es2 = es2 \\ &merge\ es1\ [\,] = es1 \end{aligned}$$

Note that *merge* compares entire events rather than just start times. This is to ensure that it is commutative, a desirable condition for some of the proofs used in Section **??**. Here is a more efficient version that will work just as well in practice:

$$\begin{aligned} &merge\ a@(e1 : es1)\ b@(e2 : es2) = \\ &\quad \textbf{if}\ eTime\ e1 < eTime\ e2\ \textbf{then}\ e1 : merge\ es1\ b \\ &\qquad\qquad\qquad\qquad \textbf{else}\ e2 : merge\ a\ es2 \\ &merge\ [\,]\ es2 = es2 \\ &merge\ es1\ [\,] = es1 \end{aligned}$$

```
data PhraseAttribute = Dyn Dynamic
                     | Tmp Tempo
                     | Art Articulation
                     | Orn Ornament
     deriving (Eq, Ord, Show)

data Dynamic = Accent Rational | Crescendo Rational | Diminuendo Rational
             | StdLoudness StdLoudness | Loudness Rational
     deriving (Eq, Ord, Show)

data StdLoudness = PPP | PP | P | MP | SF | MF | NF | FF | FFF
     deriving (Eq, Ord, Show, Enum)

data Tempo = Ritardando Rational | Accelerando Rational
     deriving (Eq, Ord, Show)

data Articulation = Staccato Rational | Legato Rational | Slurred Rational
                  | Tenuto | Marcato | Pedal | Fermata | FermataDown | Breath
                  | DownBow | UpBow | Harmonic | Pizzicato | LeftPizz
                  | BartokPizz | Swell | Wedge | Thumb | Stopped
     deriving (Eq, Ord, Show)

data Ornament = Trill | Mordent | InvMordent | DoubleMordent
              | Turn | TrilledTurn | ShortTrill
              | Arpeggio | ArpeggioUp | ArpeggioDown
              | Instruction String | Head NoteHead
     deriving (Eq, Ord, Show)

data NoteHead = DiamondHead | SquareHead | XHead | TriangleHead
              | TremoloHead | SlashHead | ArtHarmonic | NoHead
     deriving (Eq, Ord, Show)
```

Figure 6.2: Phrase Attributes

6.2 Players

Recall from Section **??** the definition of the *Control* data type:

```
data Control =
        Tempo Rational        -- scale the tempo
      | Transpose AbsPitch    -- transposition
      | Instrument InstrumentName    -- intrument label
      | Phrase [PhraseAttribute]     -- phrase attributes
      | Player PlayerName     -- player label
    deriving (Show, Eq, Ord)

type PlayerName = String
```

We mentioned, but did not define, the *PhraseAttribute* data type, shown now fully in Figure 6.2. These attributes give us great flexibility in the interpretation process, because they can be interpreted by different players in different ways. For example, how should "legato" be interpreted in a performance? Or "diminuendo?" Different players interpret things in different ways, of course, but even more fundamental is the fact that a pianist, for example, realizes legato in a way fundamentally different from the way a violinist does, because of differences in their instruments. Similarly, diminuendo on a piano and diminuendo on a harpsichord are different concepts.

With a slight stretch of the imagination, we can even consider a "notator" of a score as a kind of player: exactly how the music is rendered on the written page may be a personal, stylized process. For example, how many, and which staves should be used to notate a particular instrument?

In any case, to handle these issues, Haskore has a notion of a *player* that "knows" about differences with respect to performance and notation. A Haskore player is a 4-tuple consisting of a name and three functions: one for interpreting notes, one for phrases, and one for producing a properly notated score.

```
data Player a = MkPlayer{ pName :: PlayerName,
                          playNote :: NoteFun a,
                          interpPhrase :: PhraseFun a,
                          notatePlayer :: NotateFun a }
    deriving Show

type NoteFun a = Context a → Dur → a → Performance
type PhraseFun a = PMap a → Context a → [PhraseAttribute]
```

$$\rightarrow \mathit{Music}\ a \rightarrow (\mathit{Performance}, \mathit{DurT})$$

type *NotateFun* $a = ()$

The last line above is because notation is currently not implemented.

6.2.1 Examples of Player Construction

In order to provide the most flexibility, we define attributes for individual notes:

```
data NoteAttribute = Volume Int      -- by convention: 0=min, 100=max
                   | Fingering Int
                   | Dynamics String
                   | PFields [Float]
     deriving (Eq, Show)
```

Our goal then is to define a player for music values of type:

```
type Music1 = Music (Pitch, [NoteAttribute])
```

A "default player" called *defPlayer* (not to be confused with "deaf player"!) is defined for use when none other is specified in the score; it also functions as a base from which other players can be derived. *defPlayer* responds only to the *Volume* note attribute and to the *Accent*, *Staccato*, and *Legato* phrase attributes. It is defined in Figure 6.3. Before reading this code, recall how players are invoked by the *perform* function defined in the last section; in particular, note the calls to *playNote* and *interpPhrase*. Then note:

1. *defPlayNote* is the only function (even in the definition of *perform*) that actually generates an event. It also modifies that event based on an interpretation of each note attribute by the function *defHasHandler*.

2. *defNasHandler* only recognizes the *Volume* attribute, which it uses to set the event volume accordingly.

3. *defInterpPhrase* calls (mutually recursively) *perform* to interpret a phrase, and then modifies the result based on an interpretation of each phrase attribute by the function *defPasHandler*.

4. *defPasHandler* only recognizes the *Accent*, *Staccato*, and *Legato* phrase attributes. For each of these it uses the numeric argument as a "scaling" factor of the volume (for *Accent*) and duration (for *Staccato* and *Lagato*). Thus *Modify* (*Phrase* [*Legato* (5%4)]) m effectively increases the duration of each note in m by 25% (without changing the tempo).

```
defPlayer :: Player (Pitch, [NoteAttribute])
defPlayer = MkPlayer{ pName = "Default",
                      playNote = defPlayNote defNasHandler,
                      interpPhrase = defInterpPhrase defPasHandler,
                      notatePlayer = defNotatePlayer () }

defPlayNote :: (Context (Pitch, [a]) → a → Event → Event)
               → NoteFun (Pitch, [a])
defPlayNote nasHandler
  c@(Context cTime cPlayer cInst cDur cKey cVol) d (p, nas) =
      [foldr (nasHandler c)
             (Event{ eTime = cTime, eInst = cInst,
                     ePitch = absPitch p + cKey,
                     eDur = d * cDur, eVol = cVol,
                     pFields = [] })
             nas]

defNasHandler :: Context a → NoteAttribute → Event → Event
defNasHandler c (Volume v) ev = ev{ eVol = v }
defNasHandler c (PFields pfs) ev = ev{ pFields = pfs }
defNasHandler _ _ ev = ev

defInterpPhrase :: (PhraseAttribute → Performance → Performance)
                   → PhraseFun a
defInterpPhrase pasHandler pmap context pas m =
      let (pf, dur) = perf pmap context m
      in (foldr pasHandler pf pas, dur)

defPasHandler :: PhraseAttribute → Performance → Performance
defPasHandler (Dyn (Accent x)) =
              map (λe → e{ eVol = round (x * fromIntegral (eVol e)) })
defPasHandler (Art (Staccato x)) = map (λe → e{ eDur = x * eDur e })
defPasHandler (Art (Legato x)) = map (λe → e{ eDur = x * eDur e })
defPasHandler _ = id

defNotatePlayer :: a → ()
defNotatePlayer _ = ()
```

Figure 6.3: Definition of default Player *defPlayer*.

It should be clear that much of the code in Figure **??** can be re-used in defining a new player. For example, to define a player *weird* that interprets note attributes just like *defPlayer* but behaves differently with respect to phrase attributes, we could write:

```
weird :: Player (Pitch, [NoteAttribute])
weird = MkPlayer{ pName = "newPlayer",
                  playNote = defPlayNote defNasHandler,
                  interpPhrase = defInterpPhrase myPasHandler,
                  notatePlayer = defNotatePlayer () }
```

and then supply a suitable definition of *myPasHandler*. That definition could also re-use code, in the following sense: suppose we wish to add an interpretation for *Crescendo*, but otherwise have *myPasHandler* behave just like *defPasHandler*.

```
myPasHandler :: PhraseAttribute → Performance → Performance
myPasHandler (Dyn (Crescendo x)) pf = ...
myPasHandler pa pf = defPasHandler pa pf
```

Exercise 6.1 *Fill in the ... in the definition of myPasHandler according to the following strategy: Gradually scale the volume of each event in the performance by a factor of* 1 *through* $1 + x$*, using linear interpolation.*

Exercise 6.2 *Choose some of the other phrase attributes and provide interpretations of them, such as Diminuendo, Slurred, Trill, etc. (The trill functions from section* **??** *may be useful here.)*

Figure 6.4 defines a relatively sophisticated player called *fancyPlayer* that knows all that *defPlayer* knows, and much more. Note that *Slurred* is different from *Legato* in that it doesn't extend the duration of the *last* note(s). The behavior of *Ritardando* x can be explained as follows. We'd like to "stretch" the time of each event by a factor from 0 to x, linearly interpolated based on how far along the musical phrase the event occurs. I.e., given a start time t_0 for the first event in the phrase, total phrase duration D, and event time t, the new event time t' is given by:

$$t' = (1 + \frac{t - t_0}{D}x)(t - t_0) + t_0$$

Further, if d is the duration of the event, then the end of the event $t + d$ gets stretched to a new time t'_d given by:

$$t'_d = (1 + \frac{t + d - t_0}{D}x)(t + d - t_0) + t_0$$

The difference $t'_d - t'$ gives us the new, stretched duration d', which after simplification is:

$$d' = (1 + \frac{2(t - t_0) + d}{D}x)d$$

Accelerando behaves in exactly the same way, except that it shortens event times rather than lengthening them. And, a similar but simpler strategy explains the behaviors of *Crescendo* and *Diminuendo*.

```
fancyPlayer :: Player
fancyPlayer = MkPlayer{ pName = "Fancy",
                        playNote = defPlayNote defNasHandler,
                        interpPhrase = fancyInterpPhrase,
                        notatePlayer = defNotatePlayer () }
fancyInterpPhrase :: PhraseFun
fancyInterpPhrase pmap c [] m = perf pmap c m
fancyInterpPhrase pmap c@Context{ cTime = t, cPlayer = pl, cInst = i,
                                  cDur = dt, cKey = k, cVol = v }
                  (pa : pas) m =
  let pfd@(pf, dur) = fancyInterpPhrase pmap c pas m
      loud x = fancyInterpPhrase pmap c (Dyn (Loudness x) : pas) m
      stretch x = let t0 = eTime (head pf); r = x / dur
                      upd (e@Event{ eTime = t, eDur = d }) =
                          let dt = t - t0
                              t' = (1 + dt * r) * dt + t0
                              d' = (1 + (2 * dt + d) * r) * d
                          in e{ eTime = t', eDur = d' }
                  in (map upd pf, (1 + x) * dur)
      inflate x = let t0 = eTime (head pf); r = x / dur
                      upd (e@Event{ eTime = t, eVol = v }) =
                          e{ eVol = (1 + (t - t0) * r) * v }
                  in (map upd pf, dur)
  in case pa of
     Dyn (Accent x) -> (map (λe -> e{ eVol = x * eVol e }) pf, dur)
     Dyn PPP -> loud 40; Dyn PP -> loud 50; Dyn P -> loud 60
     Dyn MP -> loud 70; Dyn SF -> loud 80; Dyn MF -> loud 90
     Dyn NF -> loud 100; Dyn FF -> loud 110; Dyn FFF -> loud 120
     Dyn (Loudness x) -> fancyInterpPhrase pmap c{ cVol = x }pas m
     Dyn (Crescendo x) -> inflate x; Dyn (Diminuendo x) -> inflate (-x)
     Dyn (Ritardando x) -> stretch x; Dyn (Accelerando x) -> stretch (-x)
     Art (Staccato x) -> (map (λe -> e{ eDur = x * eDur e }) pf, dur)
     Art (Legato x) -> (map (λe -> e{ eDur = x * eDur e }) pf, dur)
     Art (Slurred x) ->
         let lastStartTime = foldr (λe t -> max (eTime e) t) 0 pf
             setDur e = if eTime e < lastStartTime
                        then e{ eDur = x * eDur e }
                        else e
         in (map setDur pf, dur)
     Art _ -> pfd      -- Remaining articulations:
                       -- Tenuto, Marcato, Pedal, Fermata, FermataDown
                       -- Breath, DownBow, UpBow, Harmonic, Pizzicato
                       -- LeftPizz, BartokPizz, Swell, Wedge, Thumb, Stopped
     Orn _ -> pfd      -- Remaining ornamenations:
                       -- Trill, Mordent, InvMordent, DoubleMordent, Turn
                       -- TrilledTurn, ShortTrill, Arpeggio, ArpeggioUp
                       -- ArpeggioDown, Instruction String, Head NoteHead
         -- Design Bug: To do these right we need to keep the KEY SIGNATURE
         -- around so that we can determine, for example, what the trill note is.
         -- Alternatively, provide an argument to Trill to carry this info.
```

Figure 6.4: Definition of Player *fancyPlayer*.